

# 跟我学 BASIC

苹果二型  
R 1 型 微型电子计算机辅导教材



天津市青少年科技辅导员协会 编  
天津市青少年电子计算站

類別	教
卷号	84013
日期	年 8 月 日



# 跟我学 BASIC

苹果二型  
R 1 型 微型电子计算机辅导教材

天津市青少年科技辅导员协会  
天津市青少年电子计算站

编

## 《跟我学BASIC》

天津市青少年科技辅导员协会 编  
天津市青少年电子计算站  
(科学学)杂志编辑部组织印  
天津市大港区胜利印刷厂印刷  
32开本 4印张 \* 88000字  
1984年4月第一次印刷  
印数1—20000册 定价 2.20元

# 前 言

一、本书是为中小学青少年课外开展电子计算机活动，学习使用 BASIC 语言而编写的。内容简明扼要，通俗易懂，适合于青少年科技辅导员、中学生、小学生和其他初学者学习。

二、由于不同类型的微型计算机所使用的 BASIC 语言存在一些差别，本书以 APPLE II PLUS 微型机上使用为基础，兼顾 R1（也称 LAMBDA 8300）微型机。

三、本书分三章。针对初学者的特点，贯彻理论联系实际的原则，注意上机操作，将基本 BASIC 语言和部分扩展 BASIC 语言横向分为三轮，使初学者能由浅入深逐步展开进行学习。

对于小学和初中学生，可根据他们的年龄特征和文化水平，选择本书部分内容学习。如小学生可学习第一章、第二章一至二十节以及二十二节中的随机函数、取整函数、二十七、二十八节等内容。

四、本书附有习题及参考答案，答案仅供参考，不一定是最佳程序设计。

五、本书承蒙中国科协青少年工作部张晓卫工程师、天津职业技术师范学院计算机教研室余友鸾副主任校核，特此感谢。

六、由于我们水平不高，本书可能有不少错误，欢迎大家指正。

姜肇昆 赵志复 执笔

# 目 录

前 言

第一章 电子计算机的概述 .....	( 1 )
第二章 BASIC语言和程序设计 .....	( 9 )
一、语言、程序 .....	( 9 )
二、BASIC语言学习示意图 .....	( 14 )
三、运算符号 .....	( 15 )
四、数、变量、数学表达式 .....	( 15 )
五、运算规则 .....	( 18 )
六、NEW .....	( 18 )
七、RUN 机行运行 .....	( 19 )
八、LIST 后印程序清单 .....	( 19 )
九、PRINT 打印语句 .....	( 19 )
十、LET 赋值语句 .....	( 24 )
十一、修改 .....	( 28 )
十二、FOR NEXT 循环语句 .....	( 29 )
十三、框图 .....	( 34 )
十四、TAB(X)打印格式语句 .....	( 37 )
十五 INPUT 键盘输入语句 .....	( 39 )
十六、GOTO 无条件转向语句 .....	( 40 )
十七、IF THEN 条件判断语句 .....	( 41 )
十八、子程序 GOSOB转子语句 RETURN 返回语句 .....	( 45 )

十九、READ DATA 读数置数语句 .....	( 49 )
二十、RESTORE恢复数据区语句 .....	( 51 )
二十一、函数(三角函数、反正弦、反余弦、反 正切函数 $\pi$ 、开平方根函数、随机 函数、取整函数、绝对值函数、符号 函数、对数函数、指数函数、自定义 函数 .....	( 52 )
二十二、 DIM数组说明语句 .....	( 60 )
二十三、 ON 开关语句 .....	( 65 )
二十四、 逻辑判断 .....	( 66 )
二十五、 循环的转出 .....	( 68 )
二十六、 彩色作图 .....	( 69 )
二十七、 R1机黑白作图.....	( 73 )
二十八、 R1机的音乐程序.....	( 74 )
第三章 文件存取和打印机使用 .....	( 75 )
附图一、APPLE II型机键盘图 .....	( 78 )
附图二、R1型机键盘图 .....	( 79 )
R1型机常用信息表 .....	( 80 )
APPLE II型机错误信息 .....	( 81 )
习题 .....	( 84 )
习题参考答案 .....	( 100 )

# 第一章 电子计算机的概述

## 一、电子计算机的发展概况

随着科学技术和生产的发展，于1946年世界上出现了第一台电子计算机“ENIAC”全机用了18000个电子管，6000个继电器，占地近170平方米，重约三十多吨，运算速度每秒钟可做5000次加，虽然用现在的眼光看，它有很多不足之处，但是电子计算机的诞生却不同于历史上任何机器的产生，它有条件的，部分的代替和解放了人的脑力劳动和体力劳动。从第一台电子计算机的诞生至今不足40年的历史，但它的发展却异常迅猛，今天已深入到科研、生产、军事、生活各个领域。1950年全世界只有25台电子计算机。到1970年就已有11万台。预计到1988年可达到175万台(不包括微型机)。

目前，电子计算机向两个方向发展，——巨型机和微型机。

大型机、巨型机主要用于科学计算、大型设计和大量数据处理等。

小型机、微型机主要用于企业管理，少量的数据计算、信息传送和控制等。微型机发展很快，每年差不多以100万台的数量进入市场。

我国的计算机事业是从1956年开始的，1959年研制成第一台电子管计算机，1983年又制成每秒运行亿次的巨型计算机。全国已拥有大、中、小型计算机几千台，其中80%以上都是我国自己制造的。

党中央和国务院对计算机的生产和应用是非常重视的。



国务院成立了计算机和大规模集成电路领导小组。中央领导同志说：电子计算机是一种新技术，一种完全新型的生产力。最近在国务院召开的一次座谈会上，中央领导同志说：从我国的国情出发，我认为对微型机的利用，一定要摆在重要的位置予以足够的重视。在国外，现在微型机的使用范围非常广泛。就生产领域来说，它可以用于管理，用于生产过程的控制，用于技术改造，而且效果很明显。微型机的应用，对老企业的改造，特别是对中小企业的改造，将会起到我们意想不到的作用。

无论从我国工业的现状和今后的发展方向看都需要突出抓一下微型机的应用问题。

计算机的使用不同于收音机、电视机、掌握电子计算机的人员知识越渊博，对电子计算机语言越精通，电子计算机也就能发挥出更大的效益，反之，就不能充分发挥出计算机的作用。这也就是说，要从青少年抓起，使他们尽早地接触电子计算机，学习计算机知识和技术。

## 二、电子计算机有以下几个特点：

### (一)．运算速度快

电子计算机运算速度之快，是一般人力所做不到的，巨型机每秒钟运行几亿次，相当于一个人几十年上百年的计算工作量。

### (二)．精确度高

一般计算机都可准确到七、八位数。

### (三)．具有“记忆”、逻辑判断和选择的能力。

### (四)．计算机内部的运算都是自动进行的。

### (五)．通用性强。

三、电子计算机的用途极广，主要有以下几个方面：

(一)数值计算

或称为科学计算。如我国第一台电子计算机就参加了1959年北京十大建筑的数据计算。

(二)数据处理和信息加工

用电子计算机对原始数据及时的加以记录、整理与计算，加工成人们所要求的数据形式。如1982年全国人口普查中的大量数据都是通过电子计算机进行处理的。

(三)实时控制

在工业生产、交通运输等方面的自动系统都可以用计算机控制，为实现高速度、大型化、综合化、自动化创造了条件。

(四)模拟智能

机器人的出现，机器与人下棋……等等都是计算机模拟智能的具体运用。


电子计算机用途虽然非常广泛，但它毕竟是按人给它输入的程序工作的。人是计算机的主人，而不要变成计算机的仆人。

四、电子计算机的组成及使用

各种电子计算机的设计目的不同，它的组成也有差异，这里着重介绍APPLE II微型电子计算机系统和R1微型电子计算机的主要组成部分。

(一)APPLE II微型计算机(包括主机和外围设备)。

1. 主机



中央处理器是电子计算机的核心,由它产生各种指令,并以此控制计算机的运行。CPU的振荡频率决定计算机的速度。

内存贮器是存放信息的,有许多“房间”,每个“房间”叫作一个存贮单元,每个单元有一个地址,就象房间号,每个房间中存放的数字的位数是固定的,八位微型计算机就是指每个内存单元能放一个二进制的八位数,这个数就叫作这个单元的内容。内存贮器的容量是以K为单位的,1K等于1024。APPLE II型微机内存有64K,就是 $1024 \times 64$ 个单元,这64K是随机存贮的,关机后不再保留信息。还有12K监控,自启动和BASIC扩展程序,在出厂前已存贮,关机后仍然保留信息。接口是把主机与外部设备相联结的装置。

## 2. 外部设备

### (1) 键盘 (见附图一)

通过键盘将编写好的程序输入主机。键盘上共有53个键。

①数字键:1、2、3、4、5、6、7、8、9、 $\emptyset$ ,共十个。 $\emptyset$ 是数字零。

②英文字母键:26个英文字母,与英文打字机字母排列位置相同。

③符号键与符号转换(移位)键。

在数字上方和一些键上有!、“、#、\$、%、&、’、(、)、^、\*、=、-、+、/、?、<、>、.、.、.、:、:、←、→、等符号。

当需用这些符号时,凡符号位置在按键上方的,必须先按住SHIFT移位键之后,再按符号,才能出现所需符号,

④RETURN回车键

在输入程序中需要换行,按下RETURN键,将光标移回到下一行的起始位置。

### ⑤空格键

在输入程序过程中需要留出空格位置时，要按空格键。

### ⑥RESET复位键

按复位键可使程序停止运行，恢复到执行前的状态。

其它各键的功能、使用方法将在第二章中介绍。

### (2) 显示器

可显示出输入的程序和运行的结果。显示器有彩色和黑白两种。

### (3) 磁盘驱动器

磁盘是用磁介质涂在塑料片上制成的。常用的有两种规格，即8吋和5<sup>1</sup>/<sub>4</sub>吋。

磁盘是用来存贮程序的，盘磁上有35道同心圆，又分成16个扇区，全盘可存123K。

常用的外存贮器还有硬盘、磁带、卡片、纸带等。


磁盘要求保持清洁、严禁用手接触小窗孔，用毕立即放入纸套中。存放磁盘的地方，要防磁、防潮、防热。

### (4) 打印机

能打印出显示器上的程序和运行结果。

## 3. 开关机规则

电子计算机安装完毕后应检查电源电压是否稳定、正常。电源插座要按下图装置，插头中的火线不能插错。中线与地线之间应无电压。



## (1) 开机

- ①将DOS3.3磁盘标号向上，缺口向左，慢慢送入驱动器内并轻轻关上驱动器的门；
- ②打开显示器的电源开关；
- ③打开主机电源开关，这时驱动器上红灯亮，并发出沙沙声，红灯没熄灭时不得移动磁盘；
- ④计算机系统开始自动调DOS3.3到主机内存贮器中，并在显示器上出现下列字样：

```
DOS  VERSION3.3          08/25/80
APPLE II  PLUS  OR  ROMCARD  SYSTEM
MASTER
```

```
[LOADING INTEGER INTO LANGUAGE
CARD]
```

] 这个符号叫提示符、**█**是光标。在提示符没有出现之前不要按键盘。提示符出现后，驱动器红灯熄灭，沙沙声停止，计算机才可以工作。

如主机没连接驱动器，或没装入磁盘而又需要工作时，可按 **RESET** 键，即可在屏幕上出现提示符，开始工作。

## (2) 关机

- ①在红灯不亮时打开驱动器门，取出磁盘装入专用纸套中；
- ②、关闭主机电源；
- ③、关闭显示器电源；
- ④登记使用机器的情况。

## (二) R1 即 LAMBDA 8300 微型电子计算机

R1 微型电子计算机设备比较简单，适合于青少年使用，目前国际通用的基本 BASIC 语言一般都能使用。

它的主机与键盘联在一起，可用一般电视机，做为计算机的显示器。

## 1. 键盘与使用 〈见附图二〉

### (1) 数字键与英文字母键。

与APPLE II 微型计算机相同。

### (2) 符号键与符号转换键(移位)键。

在数字键与英文字母键右上角有各种符号，基本与APPLE II 型机键相同。但有些符号位置有些变化，没有八符号，乘方改使用\*\*。

当需要符号时，必需先按住SHIFT 移位键后，再按所需符号键。

### (3) 光标反白键

按GRAPHICS键后，光标反白，在光标中出现G字。符号，再按各数字键和英文字母键时，屏幕上显示白底黑字。再按GRAPHICH键，则退出反白。

### (4) 各种小图案键。

在数字和英文字母右下角共有21个小图案、可用小图案组成各种图形。(详见第二章)

### (5) 各指令键。如PRINT、LINE NO.等详见第二章。

(6) ENTER回车键。用法与APPLE II 型机RETURN键相同。

### (7) SPACE空格键。用法与APPLE II 型机空格键相同。

### (8) RESET复位键。用法与APPLE II 型机复位键相同

## 2. 开关机规则

### (1) 开机：

①打开显示器电源开关；

②显示器屏幕亮后打开主机开关；

③在显示器左下角出现光标后，即可开始工作。

(2) 关机

- ①关主机电源开关；
- ②关显示器电源开关；
- ③登记使用机器的情况。

## 第二章 BASIC语言和程序设计

### 一. 语言、程序

从第一章的作业中的几道题，可以看出，计算机接受了人们向它发布的命令，按照人们的意图进行了工作。以这几道题为例，人们向计算机发布了什么命令呢？请将下面左右两栏对照来看：

#### 1. NEW

```
10 LET S=0
20 FOR I=1 TO 100
30 LET S=S+1
40 NEXT I
50 PRINT "S=";S
60 END
```

清除内存的程序

S是累加和，先置0

I是变量从1到100

S每次加1

执行下一个I的量

打印S值

结束

(注：R1机没有END指令。下同)

RUN

执行

#### 2. NEW

```
10 LET S=0
20 FOR X=1 TO 100
30 LET S=S+X*X
```

清除内存的程序

S是累加面积和，先置0

X是变量从1到100

S是变量X乘X累加和，X  
每次增1

```
40 NEXT X
```

执行下一个X的量

```
50 PRINT "S=";S
```

打印S值

```
60 END
```

结束

RUN

执行



3, NEW	清除内存的程序
10 FOR I=1 TO 10	变量I从1到10
20 PRINT TAB(15	打印格式打印在15-I这列
-I);	
30 FOR J=1 TO	变量J从1到2乘I减1
2*I-1	
40 PRINT “*”;	打印*号, 从左向右连打 不换行
50 NEXT J	执行下一个J的量
60 PRINT	打印空行, 换行
70 NEXT I	执行下一个I的量
80 END	结束
RUN	执行

在这里, 人们是用 BASIC 语言编制程序向计算机发布了命令。

(一). 语言: 计算机要按照人们的意图工作, 就必须使计算机懂得人的意图, 接受人向它发布的命令。但它不懂人说的语言, 就需要人用计算机的语言同它讲话、发令。

计算机语言有机器语言、汇编语言、高级语言。机器语言用二进制作代码, 用起来很困难, 难学、难写、难记、难检查、难修改, 又不通用, 于是人们又创造了高级语言。高级语言包括BASIC、FORTRAN、ALGOL、COBOL等。人们用这些高级语言给计算机发指令。计算机自己把这些语言翻译成机器指令程序去执行。人们学会了高级语言就可以很快地学会使用计算机, 而且可以不管什么机器指令, 也不必懂得计算机内部结构和工作原理, 就可以使用计算机。

高级语言还有一个好处，是可以适用于不同的计算机。

(二). BASIC语言: BASIC语言是初学者通用符号指令代码的英文缩写(Beginner's All-Purpose Symbolic Instruction Code),是一种小型会话式高级语言。但现在它已不只是初学者使用,而是一种国际通用的计算机语言了。小型机、微型机一般都配有BASIC语言。BASIC语言已经由简单的BASIC,发展成为扩展BASIC。我们学习的是APPLE II微型计算机的浮点扩展BASIC语言。

这种语言的特点是:

1. 简单易学。基本BASIC语句只有17种。符号同数学符号差不多,许多指令就同英语一样(如PRINT、END、NEW等)

2. 会话式语句(见附录信息表)。人机对话,便于查找程序中的错误。

3. 有立即执行和迟缓执行两种方式。

(1) 立即执行:不用行号,只能执行一次。如

```
PRINT 3 + 4 ↵
```

7

(2) 迟缓执行:有行号,可执行多次。如

```
10 PRINT 3 + 4
```

```
20 END
```

```
RUN
```

7


```
RUN
```

7

4. 有数据处理能力。不只是做数学运算,而且包括逻辑判断、文字、信息、情报、编写文件等。

5. 不同的计算机在使用 BASIC 语言时有小的区别，要查阅使用说明书，但基本通用。

(三). 程序。人们做事都要有步骤、进度、有次序。如今天大家来上课，从进楼门开始排程序，直到听课。



这样表达出来的步骤就是程序。这个图形是处理问题时的思维图形，在计算机程序设计里叫框图，是执行的步骤。

计算机程序是由计算机语言来表达的。编制它的过程叫程序设计。

(四). BASIC程序的构成和基本原则。

以第一章的几题为例，可以看出，BASIC语言程序有如下规定：

1. 由若干行组成；
2. 每一行叫一个语句（APPLE II型机允许在一行里写几个语句，各语句中间以：号分开）；
3. 每一个语句分别让计算机执行某个指令，做某一方面

的事。

BASIC语言程序的样式一般有三部分：

1、行号(标号、语句标号)。如10、20……。APPLE II型机的行号有1~63999，R1型机行号有1~9999。

每改行号要另起一行。即打完一行号语句，要按“回车键”(APPLE II型机的RETURN，R1机的ENTER)。

在排行号时一般按10、20、30、……排列，便于修改。

(注：R1机有LINE NO行号键，按这个键可打出相隔10的行号。)如没有按行号由小到大的顺序输入程序，在执行时，计算机可自动按由小到大的顺序执行。如两个语句先后用了同一行号，后句就把前句冲掉。

2、语句部分。行号后面的部分叫语句部分。必须有语句指令(也叫语句定义符，它规定计算机执行某个功能)及语句体(即需要执行的具体内容)

3.结束符END (注R1型机没有END指令，可省略结束符，但在必要时，可用STOP代替)

## 二、BASIC语言学习示意图

# BASIC 学习示意图

	打印作图	数据输入	算术运算函数	转移	条件判断	循环	文件存取	打印机的调用	系统调用
1	PRINT	LET	+ - * / < > < > =, =, < >, > < ^ R1机**	GOT On	IF 数学判断 THEN n R1机; IF 数学判断 THEN GOTO	FOR NEXT	LIST NEW DEL	PR #1 PR #0 R1机	
2	PRINT TAB ( ); R1机; PRINT AT X,Y;	INPUT	函数 DIM数组	GOSUB n RETURN	IF 判断算术 THEN 语句	多循环体	LOAD SAVE CATALOG DELETE	大小写	INT FP PR#6 R1机 无
3	无色作图 GR COLOR PLOT X, Y VLIN Y1, Y2 AT X HLIN X1, X2 AT Y HGR HCOLOR HPLOT TO R1机 彩色作图 PLOT 白点 GRAPHICS 作图反白	READ DATA RESTORE R1机无	DEF FN 自定义函数 R1机无	ON n GOTO ON n GOSUB RR GOTO R1机 无	IF 逻辑判断 THEN	循环 转出	INIT HELLO COPY RUN		汇编

特点：1.可以由浅入深分三轮进行。

2.把理论学习、编制程序和上机操作结合起来，可使初学者在短时间里学会。

3.对不同文化程度的学员可有不同的要求。

### 三、运算符号

代数式运算符号有：

加 减 乘 除

+ - × /

乘方

^ (R1机是\*\*)

关系式运算符号有：

等于 大于 小于 不等于

= > < <>或><

小于等于 大于等于

<= >=

括弧没有大、中、小之分，可以套用，如

( )、(( ) )、( ( ) (( ) ))

### 四、数、变量、数学表达式

#### (一)、数

1、在电子计算机里一般数的科学表示法是

S X.XXXXXXXX E S TT

符号 0~9的九位数 以10为 符 0~9二

“+”或“-”， 底的 号 位数最

“+”可不打 指数 多可到

出符号 38

如：1.234E+19 即 $1.234 \times 10^{19}$

5.876E-20 即 $5.876 \div 10^{20}$

2. APPLE II 型机和R1型机实型数的范围是 $-1 \times 10^{38}$   
 $\sim 1 \times 10^{38}$

## (二)变量

从以下几例，可以看出X值的变化：

1)10 X=1

20 PRINT "X="; X

30 END

RUN

X=1

2)10 FOR I=1 TO 5

20 X=I

30 PRINT "X="; X

40 NEXT I

50 END

RUN

X=1

X=2

X=3

X=4

X=5

3)10 FOR I=1 TO 5

20 X=X+I

30 PRINT "X="; X

40 NEXT I

50 END

RUN

X=1

X=3

X=6

X=10

X=15

变量(如以上三例中的X)是什么?从X值的变化可以看出:

1.是量;

2.值是不固定的,有的X值在执行程序的过程中可以变动;

3.占一个存贮单元;

4.变量的形式必须是英文字母开头,第二位和以后可以是字母或数字字符,叫做变量名。如A、B、AC、A1、B5、BE576等。用APPLE II型机时,变量名最多可有238个字符,但计算机只认前两位,R1型机只允许两位。

5.实型变量和字符串变量:

实型变量:代表一个实型数的变量如 $AX=3*5$

字符串变量:可以表示文字,表示符是\$。

如: A\$ = "JI SUAN JI" (计算机)

C\$ = "WANG PING" (王萍)

X\$ "C4 D4 E4 F4 G4 A4 B4" (R1机内一段音乐程序)

在字符串变量的=号右边的字符串必须加引号。

### (三) 数学表达式

计算机的数学表达式是按照计算机语言的规定,用运算符、括弧等把数、变量、函数等连接起来的数学式。



如： $5x^2 - 3x - \frac{2\sin(a)}{3}$

$5 * X \wedge 2 - 3 * X - 2 * \text{SIN}(A) / 3$

注：R1机是 $5 * X * * 2 - 3 * X - 2 * \text{SIN}(A) / 3$

## 五. 运算规则

运算要遵循下面先后顺序：

(一)、( )。没有大、中、小之分，可以套用，但必须对称，不允许失配。

(二)、函数。

(三)、乘方。

(四)、乘、除。

(五)、加、减。

(六)、同级运算要遵照先左后右的规则。

要注意以下两对数学式中①式、②式是不同的

$$\left\{ \begin{array}{l} \text{① } 23 * A / 2 + B \text{ 的算术式是 } \frac{23A}{2} + B \\ \text{② } 23 * A / (2 + B) \text{ 的算术式是 } \frac{23A}{2 + B} \end{array} \right.$$

$$\left\{ \begin{array}{l} \text{① } 23 * A / 2 * B \text{ 的算术式是 } \frac{23A}{2} \times B \\ \text{② } 23 * A / (2 * B) \text{ 的算术式是 } \frac{23A}{2B} \end{array} \right.$$

六 NEW 这条指令是要重新清除主机内存程序。

在每次输入一个新程序前，必须先输入NEW指令，将主机内存的原有程序清除。否则新程序输入后，有的原有程

序没被冲掉，将造成新旧程序交叉的混乱程序。

七 RUN 执行、运行。

这个指令是叫程序运行。

八 LIST 打印程序清单

执行这个指令可在显示器屏幕上显示出主机内存的程序。

如显示器屏幕已满，需要暂停继续显示以下程序时，可按STOP LIST键。再按STOP LIST键时，又继续显示以下程序。

注：R1机在执行LIST指令如屏幕已满时，即自行停止继续显示，出现SF信息码，停止在m(行号)语句。如需要再继续显示以下程序时可输入LISTm(行号)指令。如停止在150行，可再输入LIST 150

LIST n(行号)这个指令是要在屏幕上显示出n行语句。如LIST 90。

LIST n1, n2 这个指令是要在屏幕上显示n1行到n2行语句。如LIST 120,240。(注：R1机没有这个指令。)

九 PRINT 打印语句

在BASIC程序中，几乎每个程序都要用PRINT指令，进行打印。它的用法有以下几种：

(一) 打印数。如：

10 PRINT 3

20 PRINT

30 PRINT

```
40 PRINT 6
50 PRINT
60 PRINT 9
70 END
RUN
```

3

6

9

(二) 打印已被赋值的变量。如：

```
10 LET A=3
20 PRINT A
30 END
RUN
```

3

(三) 打印数学表达式的值。如：

```
PRINT 3*4/6
RUN
```

2

```
PRINT (3+4)*5/A
RUN
```


DIVISION BY ZERO ERROR

A没被赋值,即作0,显示出“被0除错误”。

注: R1机显示UV说明是未定义变量

#### (四) 打印标准格式。

在APPLE II型机标准格式是将屏幕分成三部分(1~16列, 17~32列, 33~40列, 见下图)打印。



```
如: 10 PRINT 1+2, 9/3, 3*4
     20 END
     RUN
```

3                      3                      12  
在第1列              在第17列              在第33列

又如:

```
10 LET A=2
20 LET B=4
30 LET C=6
40 PRINT A, B, C
50 END
RUN
```

2                      4                      6

注: R1型机标准格式是分成两部分, 0~16列; 17~31列。

#### (五) 打印字符串。

用引号把某些字符括起来，称为字符串，编入程序，由计算机全部照原样打印出来。 例如：

```
1)10 PRINT " * * * "  
20 PRINT "$ $ $ "  
30 PRINT " 3 4 3 "  
40 PRINT "MHM"  
50 END  
RUN  
* * *  
$ $ $  
3 4 3  
MHM
```

```
2)10 PRINT " * * * * * * * * "  
20 PRINT " * * * * * * * * "  
30 PRINT " * * * * * * * * "  
40 PRINT " * * * * * * * * "  
50 END  
RUN  
* * * * * * * *  
* * * * * * * *  
* * * * * * * *  
* * * * * * * *
```

```
3)10 A$ = "KE XUE"  
20 PRINT A$  
30 END  
RUN  
KE XUE
```

```

4)10 PRINT "A" , "B" , "C"
    20 PRINT "A" ; "B" ; "C"
    30 PRINT "A" ;
    40 PRINT "B" ;
    50 PRINT "C"
    60 END
RUN
A          B          C
ABC
ABC

```

；号的用法：；号使前后字符从左到右连接起来打印。

```

5) 10 LET X=3
    20 PRINT "X=" ; X
    30 END
RUN
X=3

```

```

6)10 PRINT "3 + 4 =" ; 3+4
    20 END
RUN
3 + 4 =7

```


(六)R1型微机还可以在指定位置打印字符串。

```

10 PRINT AT 12, 10; "NI HAO"
20 END

```

上题中12是行位置，10是列位置。行位置是由0到22的常数，列位置是由0至31的常数。



这里要注意：

1) PRINT X, Y, Z

与 PRINT "X", "Y", "Z"

与 PRINT "X"; "Y"; "Z"的区别。

2) 打印语句是不能赋值的, 如:

```
10 PRINT A=(X+Y+Z)/Y
```

```
10 PRINT A=2*4
```

都是错误语句

3) APPLE II型机 PRINT 有方便键可用? 代替, 就是按住SHIFT键再按?, 可代替PRINT。

R1机有PRINT键。

## 十、LET 赋值语句

LET语句是提供数据的语句之一, 是用来给程序中的变量赋值的, 可给变量赋予确定的数值。

形式: LET 变量名=数学表达式

执行这个语句的结果是在计算机的内存里开一个单元, 存放变量, 并把数学表达式的结果, 赋给这个变量, 就是把=

右边的值送给=左边的变量。好象开一个房间，住进住户。  
=不是数学里的“等于”，而是借用=符号，好象一座“桥”  
把河右边的住户过渡到左边的房间里。过这座“桥”的，只  
许从右往左走。

例题： 1)	2)
10 LET X=3	10 LET X=3
20 LET Y=4	20 LET Y=4
30 LET X=Y	30 LET Y=X
40 PRINT X, Y	40 PRINT X, Y
50 END	50 END
RUN	RUN
4            4	3            3

3)

```
10 LET X=3
20 PRINT X
30 LET X=X+1
40 PRINT X
50 END
RUN
3
4
```

从以上例题可以看出=号不是数学里的“等于”，而且，  
从例题3)可看出 $X=X+1$ 在计算机程序里是正确的，而在  
代数式里则是错误的。

LET在使用APPLE II型机和R1型机时可省略。

题例4)

```
5 A=0 : B=0 : C=0 : D=0 : E=0
```



```

10 B=12
20 C=A+B+3
30 D=A+B-4
40 E=A
50 PRINT A,B,C,D,E
60 A=6
70 E=A
80 PRINT A,B,C,D,E
90 END

```

RUN

0	12	15
8	0	
6	12	15
8	6	

R1机执行结果(R1机第5行号语句要分成5个行号写):

0	12
15	8
0	
6	12
15	8
6	

在执行这个程序时，各语句中各变量的变化情况如下：

变量 语句	A	B	C	D	E
5	0	0	0	0	0
10	0	12	0	0	0
20	0	12	15	0	0
30	0	12	15	8	0
40	0	12	15	8	0
50	0	12	15	8	0
60	6	12	15	8	0
70	6	12	15	8	6
80	6	12	15	8	6

对于LET语句还要说明几点：

- 1、一个变量赋值给其它变量后自己值不变。
- 2、一个变量被赋值后，原值被新值代替，不再保留。
- 3、变量如没赋值，按0对待，但R1型机没赋值就在显示器上显示UV信息码。

4、=两边如颠倒互换，语句的意义也改变。如 $X=Y$ 改成 $Y=X$ 时则意义不同。

5、每个语句中不允许超过一个=符号。如：  
 $X=3 * Y+2=(Y+1) * (Y+2)$ 是错误语句。

6、不允许向表达式赋值。如： $X+Y+Z=1+2+3$ 是错误语句。

## 十一、修改

APPLE II 型机修改方法有：

1. 改单个字符：将光标移在修改的字符上输入新字符就可将旧字符改掉。

2. 修改一行语句：输入行号按回车键，就删除这行语句。或输入行号再输入新语句，就冲掉原行号语句。

3. 使用DEL n1, n2 删除n1行号到n2行号之间的语句。重新LIST才能删除内存中的程序。

4. 使用ESC键修改：在一个程序经过LIST列出清单以后使用ESC键。步骤是：

按一下ESC键；

按I键，使光标上移到要修改的语句行号处；

按J键，将光标左移到本行第一列；

按→右移键，将光标移到要修改的字符处进行修改；

修改后再用→右移键将光标移到本语句最后一个字符右边；

按回车键将光标移出本程序再输入LIST指令检查程序修改情况。

此外，按ESC键后，再按K键可让光标右移，按M键可将光标下移。

R1型机修改方法：

1. 改单个字符：将光标移在要修改字符右一列，按EDLETE键，将要修改字符删除，再打上新的字符。

2. 修改一行语句：与APPLE II 型机相同。

3. 使用EDIT键，将要修改的语句拉到屏幕下方修改。步骤是：

按↑上行键或↓下行键将光标移到要修改的语句行号

上。

按EDIT键把光标所在的那行语句拉到屏幕下方；

用修改单个字符方法进行修改；

修改后，按ENTER回车键将改完的语句送回屏幕上方。

4. 要想删除正在打的程序，可按BREAK键，将全语句取消。

## 十二、 FOR NEXT 循环语句

### (一)、 单层循环。从三个例题谈起

例一是习题十六，如果只用PRINT语句，打印出那个图形是很麻烦的，我们可以设计另一个程序，就比较省事：

```
10  FOR I=1 TO 10 STEP 1
20  PRINT  "*" ;
30  NEXT I
40  PRINT
50  FOR I=1 TO 10 STEP 1
60  PRINT TAB(10) ; "*"
70  NEXT I
80  FOR I=11 TO 20 STEP 1
90  PRINT TAB(I) ; "*"
100 NEXT I
110 END
```

还可以加上：

```
25  FOR U=1 TO 300
26  NEXT U
```

(注：在R1机第60语句改为PRINT TAB(9)；"\*"。

第80语句改为FOR I=10 TO 19 STEP 1.第25语句改为FORU=1 TO 100.各语句中的“\*”号可改为“X”)

打印出图形好象一个小虫子在石头上很慢地爬、爬到边上很快地垂直掉下去，又很快地从斜坡上滚下去。这里用了循环语句。

例二是要计算20！即 $1 \times 2 \times 3 \cdots \times 20 = ?$

```
10   LET  X=1
20   FOR  I=1 TO 20 STEP 1
30   X=X*I
40   NEXT I
50   PRINT "X=",X
60   END
```

RUN

X=2.43290201E+18

这里也使用了循环语句，比用计算器速度要快得多。

例三是，相传古代印度国王舍罕要奖赏他聪明能干的宰相达依尔，国王问达依尔你要什么，达依尔说：我只要国王在国际象棋棋盘的第一个格里放一粒麦子，第二个格里放两粒，第三个格里放四粒，以后依次每格加一倍。一直放到棋盘最后一格，第六十四格，把这些麦子奖给我，我就感恩不尽了！国王想：这还不容易！就叫人扛来一袋小麦，但不一会就用完了，再扛来一袋，很快又用完了。国王很奇怪，算不清这笔帐。我们用电子计算机来计算一下，即求S（麦粒数之和） $= 1 + 2 + 2^2 + 2^3 + \cdots + 2^{63}$

```
10   S=0           S是麦粒数之和，先置零。
20   I=0           变量I是指数，指数从0开始即20
30   FOR  I=0 TO 63  指数从零到63  即
```

$$2^0 + 2^1 + 2^2 + 2^3 + \dots + 2^{63}$$

```

40   P=2^I
50   S=S+P
60   NEXT I
70   PRINT "S=" , S
80   M=S/(1.42E+8)    1立方米小麦约有
                        1.42×108粒
90   PRINT "M" , M
100  END
RUN

```

S=1.84467141E+19

粒

M=1.29906648E+11

立方米

R1机第40语句改为P=2\*\*I

从以上三例可以看出循环语句的用处是很大的。

形式:FOR 循环变量=初值 TO 终值 STEP步长

### NEXT 循环变量

初值、终值、步长可以是数,也可以是已被赋值的变量。

说明:

1.循环体:从FOR到与之相应的NEXT两条语句当中的语句叫做一个循环。如例三的40、50语句

2.循环变量与循环体中语句有两种关系:

一是循环变量在循环体中语句里出现,是循环体里变量和数学式的关系,如例二的I。

二是循环变量在循环体中其它语句里不出现,只是控制循环体里其它语句执行的次数。如例一10~30句的I。

3.STEP步长是1时,可省略,隐含在语句中。如STEP

步长是0时，循环变量不变。STEP步长可以是负数，但必须符合初值到终值的步长规律。如 FOR I=10 TO 1 STEP -1。

4. 退出循环体。当环变量越出终值时，就退出循环体，开始执行NEXT下一行语句。

(二) 双重循环。有编制程序时，有不少问题仅靠单层循环是不能解决或比较烦麻，就要用双重循环，组成循环嵌套例如打印图形，

```
*****
*****
*****
*****
```

如用单层循环程序是：

```
10 FOR I=1 TO 4
20 PRINT "*****"
30 NEXT I
40 END
```

改用双重循环程序是：

```
10 FOR I=1 TO 4  变量I代表行
20 FOR J=1 TO 9  数，从1行到4行
30 PRINT "*" ;  变量J代表每行
40 NEXT J        " "号个数，从
50 PRINT        1个到9个。
60 NEXT I
70 END
```

10语句至60语句是外循环，20语句至40语句是内循环，当变量值是J时，变量J值要从1变到9，30语句要执行9次，连

续横排打印出九个\*号。当J越出终值退出内循环体执行到50语句时，仍连接第九个\*号打一空行。然后在执行60语句时要另起一行，而变量I值从1变到4。每增加1，内循环就要再执行一次，J重新取新值，从1变到9，打印九个\*号。当I增值越出终值4时，退出外循环，结束。

又如打印图形：

```
*  
* *  
* * *  
* * * *
```

程序是：

```
10   FOR I=1 TO 4  
20   FOR J=1 TO I  
30   PRINT "*" ;  
40   NEXT J  
50   PRINT  
60   NEXT I  
70   END
```

说明：

1. 在双重循环里每重循环的FOR变量和NEXT变量必须相对应。
2. 循环嵌套不得交叉。如：



正 确

```

FOR I
  FOR J
  ...
NEXT J
NEXT I

```

错 误

```


FOR I
  FOR J
  ...
NEXT I
NEXT J

```


### 十三、框图(流程图)

由于程序当中有循环语句以及GOTO(无条件转向语句)、IF—THEN(条件语句)和GOSUB(转子语句)等语句,使得有些程序在执行时不完全是按行号大小顺序执行,往往容易搞不清程序走向.在编制程序时,特别是一些较长、较复杂的程序,也往往需要先借助一种图形来理清设计程序的思路,这种图形就叫做框图,也叫流程图。


(一)常用的框有:



开始或结束



矩形框、处理框。




菱形框、判断框、检查框。  
可有满足条件(YES,是)和不满足条件(NO,否)两个出口。


(二) ↓ 表示程序执行流向。

(三) 一个框表示一段程序(一个语句或几个语句)的功能,要简明。

(四) 以第十二节单层循环例二说明框图的使用。



以双重循环例题说明框图的使用：



## 五)框图的用法步骤:

1. 确定计算方案;
2. 画框图并检查无误;
3. 参考框图编制程序;
4. 调试运行程序, 如有错误可对照框图检查。

## 十四、TAB(X) 打印格式语句

TAB(X)是打印格式语句, 可将字符打印在 X 列上, 用来打印表格或打印图形。

形式:

```
PRINT TAB(X); A
```

X是数、数学式,代表第几列,一般是1~40(列)之间的数(打印机可打1~80列), R1机是0~31(列)。大于这个数时,即自动换到第二行,继续打印。

A可以是数、赋值的变量、数学表达式或字符串。

；处必须用；号。如用，号就打印在屏幕中间。

```
例如: 10 PRINT TAB(5); "*"
       20 PRINT TAB(10); "*"
       30 PRINT TAB(20); "*"
       RUN
```

```
           *
            *
             *
10 PRINT TAB(5); "*"
20 PRINT TAB(10); "*"
30 PRINT TAB(20); "*"
RUN
      *   *   *
```

说明:

1. X指的是开始位置, 包括符号。 如

```
10   A=12
20   B=-12
30   PRINT  TAB(5);A
40   PRINT  TAB(5);B
50   PRINT  TAB(5);"12"
60   END
RUN
```

```
1 2 3 4 5 6 7 8 9 10
1 |       1 2
2 |       - 1 2
3 |       1 2
```

2, X必须有确切值, 否则作零对待; X不能是负数, X如有小数, 即自动取整(如 2.4即取2)。

3, 一个语句使用多个TAB时, X值只能从左到右顺序增加, 不能减少。

如: 正确的: TAB(5);"A", TAB(10);"B",  
TAB(20);"C"

错误的: TAB(20);"A", TAB(10);"B",  
TAB(5);"C"

注: R1机TAB(X);"A"后如果是;号(即连续横排打印), 而这语句又在循环体里时, 要在X后加上循环变量。

如: 打印

```
* * * * *
* * * * *
* * * * *
* * * * *
```

第一个\*要打印至第10列，程序是，

```
10 FOR I=1TO 4
20 FORJ=1TO 5
30 PRINT TAB(10+J-1);“*”;
40 NEXT J
50 PRINT
60 NEXT I
```

### 十五、INPUT 键盘输入语句

INPUT 语句也是提供数据的语句之一，可以在程序运行后，用键盘把数值输入给相应的变量。可给变量赋不同的值，也可以给几个变量赋值。

- 形式：① INPUT X  
② INPUT X, Y, Z……  
③ INFUT “字符”；X

注：R1机没有第②③种形式。X、Y、Z是数字变量、字符串变量。

说明：

1. 计算机执行到 INPUT语句时，屏幕上出现一个？号（R1机是出现光标），再由计算机使用者用键盘输入数据，输入后再按回车键，把这个数值赋给相应的变量，然后计算机就连续执行下面的语句。

2. 在运行完一次程序后，如需要INPUT另外数值时，不需要新写程序，只要使程序重新运行一次，出现？号时，再输入另外数值。

3. 输入数值必须是常数，不允许是函数或表达式。

4. 用一个INPUT语句需输入若干变量值时,各变量中间用,号分开,各数值中间也要用,号分开。如果输入数值不足变量数时,显示器上就仍会显示出?号,须继续输入数值。

例如,要编一个程序给全班50名学生分别计算语文、数学、英语、自然等四门课程的平均成绩,并在每人学号后打印出来。

```
5. FOR I=1 TO 50
10 P=0                平均成绩P先置0
20 INPUT "Y=";Y       输入语文分
30 INPUT "S=";S       输入数学分
40 INPUT "E=";E       输入英语分
50 INPUT "Z=";Z       输入自然分
60 P=(Y+S+E+Z)/4
70 PRINT "NO. ";I,P
80 NEXT I
90 END
```

每循环运行一次,即用键盘输入一个学生的四门课分数,分别计算打印。

#### 十六、GOTO 无条件转向语句

用来无条件地改变程序的执行顺序。好象是一个指路标牌,指示程序的走向。

形式: GOTO n

n是行号

如果要打印

```
*****
$$$$$
```

```

# # # # #
* * * * *
$ $ $ $ $
# # # # #
* * * * *
$ $ $ $ $
# # # # #
* * * * *
$ $ $ $ $
* * * * *

: : : : :
: : : : :
: : : : :
: : : : :
: : : : :

```

可编程序

```

10 PRINT " * * * * *"
20 PRINT " $ $ $ $ $"
30 PRINT " # # # # #"
40 GOTO 10
50 END

```

可连续执行。

如需中断执行可按 **STOP** 键。  
**RUN**

注：R1机执行到屏满时停止。如需继续显示，可输入  
CONT。

十七、IF THEN 条件判断语句

在第三十二习题里，小明想用“\*”号打印两条交叉斜线，但他编制的程序是；



```

10 FOR A=1 TO 21
20 PRINT TAB(5+A); '*";TAB(26-A);
  "*"
30 NEXT A
40 END

```

执行这个程序不能出现予想的结果，这是因为在程序执行过程中， $(5+A)$ 的值会逐渐增大， $(26-A)$ 的值会逐渐减小，当 $(5+A)$ 大于 $(26-A)$ 时，第20语句就违反了打印格式语句的规定，会改变斜线的走向。

要想得到予想的结果，就要在 $(5+A)$ 大于 $(26-A)$ 时把第20语句变成PRINT TAB(26-A)"\*";TAB(5+A); "\*"先打印 $(26-A)$ 这列的\*号。这个程序可以使用IF THEN 语句，即，

```

10 FOR A=1 TO 21
15 IF(5+A)>(26-A) THEN 25
20 PRINT TAB(5+A);"*";TAB(26-A);"*"
22 GOTO 30
25 PRINT TAB(26-A);"*";TAB(5+A);"*"
30 NEXT A
40 END

```

就是增加了第15、22、25三个语句。

第15语句起了判断的作用。

注：R1机第15语句应是

```
IF(5+A)>(26-A) THEN GOTO 25
```

判断语句是有条件转向语句。

形式 IF 关系判断式 THEN n


注：R1机是：IF 关系判断式 THEN GOTO n

## n是行号。

这条语句的意思是，如果关系判断式的条件成立，就转移执行第八行语句。如果条件不成立，就继续执行本语句行的下一行语句，

在使框用图设计条件判断语句时，要用菱形框。

执行本行语句的下一行语句



执行本行语句的下一行语句

关系判断式可以是：

1. 一个变量与一个变量比较，如  $X > Y$ ；
2. 一个变量与一个数比较，如  $X < 5$ ；
3. 一个变量与一个数学表达式比较。如  $X \geq A * 5$ ；
4. 两个表达式比较。如  $5 * A * Y < B * 8$ 。

例一：列举一批数，打印出大于零的数。

```
10 INPUT N
20 IF N > 0 THEN 40
30 GOTO 10
40 PRINT N
50 GOTO 10
```

60 END

例二：用条件判断语句打印以下图形，

\*

\* \*

\* \* \*

\* \* \* \*

10 FOR I=1 TO 4

20 A=I

30 PRINT "\*";

40 A=A-1

50 IF A=0 THEN 70

60 GOTO 30

70 PRINT

80 NEXT I

90 END

例三：计算20! 并画框图

10 I=0

20 X=1

30 I=I+1

40 X=X\*I

50 IF I<20 THEN 30

60 PRINT "X=",X

70 END

I是变量表示阶乘数


X是变量表示阶乘的积

每执行一次I增值1

X乘以变量I

30 I小于20时要转向30语

句使变量I和X增值



## 十八. 子程序

GOSUB	转子语句
RETURN	返回语句

在一个程序里，如果要多次进行某种运算或多次执行某个、某几个语句，就要重复书写这些语句，很烦琐，也容易出错，就可以运用子程序的方法。

形式： GOSUB n

⋮

RETURN

n 是行号

如打印

1	1	1	1
#	#	#	#
2	2	2	2
#	#	#	#
3	3	3	3
#	#	#	#
⋮	⋮	⋮	⋮
⋮	⋮	⋮	⋮
9	9	9	9
#	#	#	#

程序是

主 程 序	⎧	10 FOR I=1 TO 9
		20 PRINT I; I; I; I
		30 GOSUB 500
		40 NEXT I
		50 END

子 程 序	⎧	500 PRINT "####"
		510 RETURN

注：R1 机 要用 STOP 代替 END

说明：

1. 把多次重复使用的程序编成一个程序就是子程序，如上题500~510语句；调用子程序的程序就是主程序，如上题10~50语句。


2. 主程序调子程序时用GOSUB n (行号) 语句。子程序行号必须排在主程序结束符行号以后，但行号不必与主程序紧紧相连。子程序在结束时要有 RETURN语句。

3. 子程序执行完时，就转回到GOSUB 调用子程序那行语句下面的语句，继续执行，GOSUB与RETURN必须对应使用。

4. 主程序不允许插入子程序中间。

5. 转入子程序的次数没有限制。

6. 子程序可以嵌套子程序。不同机型对能嵌套几重子程序的数量有不同限制。如：



7. 一个子程序只能有一个入口，但可以有多出口。如子程序里有几个IF语句，就有多出口。

例一、打印

图形，可编带有子程序的程序。

```
      *
      ***
     *****
      ***
      *

10  FOR I=0  TO  2
20  GOSUB  500
30  NEXT I
40  FOR I=1  TO  0  STEP -1
50  GOSUB  500
60  NEXT I
70  END

500  FOR J=-I  TO  I
510  PRINT  TAB(10-I); " * ";
520  NEXT J
530  PRINT
540  RETURN
```

例二、求  $S=A_1+B_1+C_1$  A、B、C、用键盘输入

```
10  INPUT  A
20  P=A
30  GOSUB  500
40  X=R
50  INPUT  B
60  P=B
70  GOSUB  500
80  Y=R
```

R是阶乘积

```

90 INPUT C
100 P=C
110 GOSUB 500
120 Z=R
130 S=X+Y+Z
140 PRINT "S=";A;"1+";B;"1+";C;"1=";S
150 END
500 R=1
510 FOR I=1 TO P
520 R=R*I
530 NEXT I
540 RETURN

```

### 十九、 READ DATA 读数置数语句

注：R1机没有这个语句。

先看一个程序：

```

10 A=35
20 B=37
30 C=41
40 D=58
50 E=80
60 S=A+B+C+D+E
70 PRINT S
80 END

```

如果变量和数据再多，就会很麻烦，也容易出错，可以用 READ DATA 语句解决这个问题。如把上一程序改成：

```

10 READ A, B, C, D, E

```



```

20   S=A+B+C+D+E
30   PRINT S
40   DATA 35, 37, 41, 58, 80
50   END

```

形式    READ    变量  
          ⋮  
          DATA

说明：

1. READ是执行语句，而DATA是非执行语句，只是为给READ语句准备数据，执行 READ时从DATA语句中顺序取出数据，依次给READ语句中变量赋值。

2. 变量和数据的顺序要相对应。变量间和数据间以，号分开，在最后一个变量和数据后不加，号。

3. DATA数据个数要等于或大于READ变量数。

4. 数据表有一个“暗指针”，每读完一数，指针就右移至下一个数据，再读数时就读这个数据。

5. DATA可放在程序内任何位置，不一定紧靠READ语句之后。

6. DATA只能置常数，不能置变量或表达式。

7. 一个READ和一个DATA语句，可以分成两个或两个以上的READ和DATA语句。如前面那个程序可写成：

```

10   READ A, B
20   READ C, D, E

```

```

30   S=A+B+C+D+E
40   PRINT  S
50   DATA  35
60   DATA  37, 41, 58
70   DATA  80
80   END

```

例题： 求下列N组数的平均值

- ① 35, 37, 41, 58
- ② 62, 71, 81, 95, 103
- ③ 256, 301, 203

```

10  INPUT  N
20  S=0
30  FOR I=1 TO N
40  READ  A
50  S=S+A
60  NEXT  I
70  Y=S/N
80  DATA  .....
90  PRINT  "Y=" ; Y
100 END

```

在求第一组数平均值时，将 4 用键盘输入N，DATA 后置第一组数。求第二组，第三组数平均值时，分别将 5、3 用键盘输入N，DATA 后分别置第二、三组数。

## 二十、RESTORE 恢复数据区语句

形式： n RESTORE n是行号

执行这个语句时，把数据表中的“暗指针”移回到数据表

开头，使DATA语句中数据从头开始，但不是使程序回到第一个READ语句，也不是从READ语句第一变量重新读起。

例：

```
10 READ A,B,C,D,E
20 PRINT A,B,C,D,E,
30 RESTORE
40 READ F,G
50 RESTORE
60 READ I,J,M,N
70 PRINT F,G,I,J,M,N
80 DATA 30,45,60,80,100
90 END
```

RUN

30	45	60
80	100	
30	45	30
45	60	80

## 二十一、 函 数

在BASIC语言中还有一些标准函数，可以直接引用。例如：三角函数、开平方根函数、随机函数、取整函数等等，现分别介绍如下：

### (一)三角函数

形式： SIN(X)  
          COS(X)  
          TAN(X)


X为数学表达式、变量。应是弧度。

由角度换算成弧度的关系是：

$$X * \pi / 180$$

在R1机上设有SIN、COS、TAN按键。在APPLE I型机上需通过键盘输入。

例题：雷达发出电波，1/3750秒收到飞机回波，已知雷达的仰角是 $22^\circ$ ，电波速度是 $3 \times 10^8$ 米/秒，问飞机高度是多少米？



$$H = ? \quad S = 1/3750 \times 1/2 \times 3 \times 10^8$$

$$H = S \times \sin(22 \times \pi / 180)$$

编写程序：

```
10 S=1/3750*1/2*3*10^8
20 H=S*SIN(22*3.14159/180)
30 PRINT "H=" ;H
40 END
```

注：R1机第10语句改为：

```
10 S=1/3750*1/2*3*10**8
```

(二)反正弦、反余弦、反正切函数

在R1机上除有求三角函数值按键外，还有求反正弦、反余弦、反正切函数键，由函数值求出度数。

形式:  $ASN(X)$

$ACS(X)$

$ATN(X)$

得到的角度在  $-\frac{\pi}{2} \sim \frac{\pi}{2}$  之间。

$x$  为数学表达式或变量。

$x$  值是弧度, 求度数时应注意换算。

例如: 求反正切函数 1 是多少度?

```
PRINT ATN1/(3.14159/180)
```

45

(三).  $\pi$

R1 机上有  $\pi$  按键, 在显示器上以 PI 表示, 准确到 3.141592。

在 APPLE II 型机上没有  $\pi$  键, 用时需用键盘输入数据。

(四) 开平方根函数

形式:  $SQR(X)$

$X$  为数学表达式或变量, 但  $X > 0$ , 不能用负数。

例 1, 

```
PRINT SQR(36)
```

6

例 2, 解  $ax^2 + bx + c = 0$

由一元二次方程求根公式知道:

$$x_{1,2} = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}$$

编写程序:

```
10 INPUT A
```

```
20 INPUT B
```

```
30 INPUT C
```

```
40 Z=B^2-4*A*C
```

```

50 IF Z<0 THEN 120
60 Z=SQR(Z)
70 X1=(-B+Z)/(2*A)
80 X2=(-B-Z)/(2*A)
90 PRINT "X1=";X1
100 PRINT "X2=";X2
110 GOTO 130
120 PRINT"WU SHI GEN"
130 END

```

### (五)随机函数

随机函数是开机后，有一振荡源，经分频后，随机函数发生器产生很多组数，叫随机函数，这组数没有一定规律，是人们不能掌握的数。

形式： $RND(X)$   $X$ 是常数。

说明：

1.在 $X>0$ 时每次得到一个新的随机数。

在 $X\leq 0$ 时每次得到一个与前次相同的随机数。

R1机设有RND键，输入RND后每次就产生不同的随机数，

2.随机数的值都在大于0小于1之间。

例1.打印不同的随机数。

```

10 PRINT RND(1)
20 GOTO 10
30 END

```

注：R1机第10语句改为PRINT RND

例2.在屏幕上打印出无规律的满天星（可用\*号）。

```

10 FOR I=2 TO 20

```

```

20 G=RND(5)*40
30 PRINT TAB(G); "*"
40 NEXT I
50 END

```

注：R1机第20语句改为 $G=RND*31$

### (六)取整函数

形式：INT(X)

X是算术表达或变量。

说明：

1.取整函数得到的数值是小于或等于X值的最大整数。

例1. PRINT INT(3.2)

3

例2. PRINT INT(-3.2)

-4

例3.10 INPUT A

20 INPUT B

30 X=INT(A+B)

40 PRINT "A+B=", A+B

50 PRIN"INT(A+B)=", X

60 GOTO 10

70 END

RUN

? 3.2

? 5.3

A+B=8.5

INT(A+B)=8

? -3.2

? -5.3

A + B = -8.5

INT(A + B) = -9

?

2. 取整函数，只取整数，如要小数“四舍五入”可在 INT ( )里的数值上加上0.5。

例1. PRINT INT(9.6+0.5)

10

例2 PRINT INT(9.4+0.5)

9

3. 想要在小数点后保留两位，(例如计算人民币的元、角、分，只保留到分)，可以用下面的办法，

10 A=115.6364

20 X=INT(A\*100+0.5)/100

30 PRINT X

40 END

RUN

115.64

4. 如果要在小数点后保留到三位或五位，用什么办法，请大家自己去想。

5. 把取整函数和随机函数联合使用可以给小学生出一些数学题。如给小学生出几道每次数值不同的百位以内正整数加法题，如答错了，可再答。答对了，就再出一道。程序是：

10 A=INT(RND(5)\*100)

20 B=INT(RND(5)\*100)

30 PRINT A; "+"; B; "=?"



```

40 INPUT X
50 C=A+B
60 IF X<>C THEN 40
70 PRINT "GOOD!"
80 GOTO 10
90 END

```

R1机第10、20语句改为：

```

10 A=INT(RND*100)
20 B=INT(RND*100)

```

### (七). 绝对值函数

形式： $ABS(X)$  X可是算术表达式

或变量。如同初等代数中求  $|X|$  的值一样。

例如： $PRINT ABS(-3.2)$   
3.2

### (八). 符号函数

形式： $SGN(X)$

当  $X > 0$  时  $SGN(2) = 1$

当  $X < 0$  时  $SGN(-2) = -1$

当  $X = 0$  时  $SGN(0) = 0$

### (九). 对数函数

形式： $LOG(X)$

指X的自然对数。

例如： $LOG(5)$ 就是求以e为底的对数 $\ln 5$ 。

$$e = 2.71828$$

以e为底的对数和以10为底的对数换算关系是：

$$\text{LOG}_{10}X = \frac{\text{LOG}e^x}{\text{LOG}e^{10}}$$

### (十) 指数函数

形式: EXP(X)

以e为底的X次方。

例如: EXP(2)就是求 $e^2 = ?$

在APPLE II型机X要小于88, 否则要溢出。

### (十一) 自定义函数

在BASIC语言中除应用前面讲到的标准函数外, 还允许使用者自己定义函数, 为我们结合本身情况编写程序提供了方便。

形式: DEF FN函数名(实型变量) = 数学表达式 1

⋮

FN 函数名(数学式2)

函数名可任选用26个英文字母中任意一个。

括弧中的实型变量, 是一个虚设的变量, 代表所定义函数的自变量。

括弧中的数学式2可以是常数或表达式, 也可以是已赋值的变量。

数学式1是自己定义的函数形式。这就是用FN函数名(实型变量)这个语句指令来代表一个较复杂的数学表达式

例一、做一个实验, 发现有以下规律:

$$y = 3x^2 + 6x + 5$$

设x分别等于2, 4, 6, 8, 10, ……20求出不同x值对应的y值。

编写出程序是;

```

10 DEF FN Y(X)=3 * X^2+6 * X+5
20 FOR X=2 TO 20 STEP 2
30 PRINT X, FN Y(X)
40 NEXT X
50 END

```

例二、打印出半径分别是1、2、3、……10厘米时的圆面积。

```

程序是:10 DEF FN W(R)=3.14159 * ^R2
20 FOR R=1 TO 10
30 PRINT "R="; R, "W="; FN W (R)
40 NEXT R
50 END

```

注：R1机中没有自定义函数

## 二十二、DIM 数组说明语句

DIM是数组说明语句的语句指令。执行DIM语句，在计算机内存中开辟予定的存贮单元，以留给数组的各元素存放数值。一维数组可存放7248个元素，多维数组受内存的限制。

### (一) 下标变量，

形式： $A(X)$        $A(X, Y)$

X、Y是下标，是0或正整数；也可以是有确切值的简单变量或数学表达式。

这是一种可以使程序简化的变量。

例如：求五个没有规律的数的和。不能用循环语句，只能用读数置数语句，程序是：

```

10 READ A1, A2, A3, A4, A5
20 H=A1+A2+A3+A4+A5

```

```
30 PRINT "H=" ; H
40 DATA 2, 5, 6, 3.1, 8.7
50 END
```

如改用下标变量，程序是：

```
10 H=0
20 FOR I=1 TO 5
30 READ A(I)
40 H=H+A(I)
50 NEXT I
60 PRINT "H= ";H
70 DATA 2, 5, 6, 3.1, 8.7
80 END
```

(二). 数组。同一个字母组成的那些下标变量，称为数组，好象一个小队。如A(1)、A(2)、A(3)……等，叫做A数组。下标变量叫做数组元素。好象一个个队员。每个下标变量(数组元素)在计算机里占一个存贮单元。

有一个下标的数组，称为一维数组，如A(I)；两个下标的数组称为二维数组，如A(I, J)。

(三). 一维数组说明语句。

形式： DIM变量(下标)

前面求五个没有规律的数的和那道题，程序还比较简单，如果要改成求100个没有规律的数的和，就十分麻烦。但用一维数组定义语句设计，就比较简单，程序是：

```
5 DIM A(100)
10 H=0
20 FOR I=1 TO 100
30 READ A(I)
```

```

40 H=H+A(I)
50 NEXT I
60 PRINT "H=" ; H
70 DATA 2, 5, 6, 3.1, 8.7, .....(一百个数)
80 END

```

注：R1机可将30语句改为INPUT A(I)删除第70、80语句。

说明：

1.在执行到第5语句时，计算机就给A数组留出101个存储单元(下标是从0开始到100)。因此，数组说明语句必须放在数组元素使用之前。

2.数组元素少于11个时可不定义，即在程序当中如果没有数组说明语句时，就被认为是DIM变量(10)数组的下标只能用0~10之间的数，也就是只给数组空11个单元存放元素。

注：R1机数组元素少于11个时也必须定义。

3.在同一程序中需要使用n个数组时，可以同时写明几个数组，中间以,号分开。如DIM A(30),B(10),E(15)

4.下标变量也可以以字符串的形式出现，如要打印全班40个学生的姓名可以编程序：

```

10 DIM A$(40)
20 FOR I=1 TO 40
30 READ A$(I)
40 NEXT I
50 FOR I=1 TO 40
60 PRINT A$(I)
70 NEXT I
80 DATA "LI HONG", "WU PING",

```

……(四十个人姓名)

90 END

如打算打印100张,可以再嵌套一个循环语句:

```
45     FOR   J=1   TO   100
75     NEXT  J
```

如果要用标准格式打印姓名、语文成绩、数学成绩,可用三个数组说明语句:

```
10  DIM A$(40),Y(40),S(40)
20  FOR I=1 TO 40
30  READ A$(I),Y(I),S(I)
40  PRINT A$(I),Y(I),S(I)
50  NEXT I
60  DATA "LIHONG",97,99,"WUPING",96,
      100,……(四十个人姓名及成绩)
70  END
```

(四)二维数组说明语句:

带有两个下标的变量,称为双下标变量,如A(X,Y)。由双下标变量组成的数组,叫做二维数组。二维数组说明语句表明所用数组是二维的。

形式: DIM 变量(下标,下标)

例如:某学校各年级、各班学生人数如下表:

年级/班级	1班	2班	3班	4班
1年级	36	40	38	39
2年级	38	37	39	42
3年级	36	35	34	33
4年级	39	32	33	30
5年级	39	37	35	0

当讲到各年级各班人数时，我们可以说一年一班是36人  
 一年二班是40人  
 一年三班是38人  
 一年四班是39人

为简便起见，也可改写成：

36(1,1) 40(1,2) 38(1,3) 39(1,4) 全校各班

都可以这样写出来。

如要求打印出各班、各年级和全校的总人数，可用二维数组说明语句编写如下程序：

```

10   G=0                               G是全校总人数
20   DIM   A(6,4)
30   FOR   I=1   TO 6
40   H=0                               H是年级人数
50   FOR   J=I   TO 4
60   READ  A(I,J);
70   PRINT TAB(8*J);A(I,J);
80   H=H+A(I,J)
90   NEXT  J
100  PRINT
110  PRINT  H
120  G=G+H
130  NEXT  I
140  PRINT  G
150  DATA 36,40,38,39,38,37,39,42,36,
          35,34,33,39,...32,33,30
160  DATA 39,37,35,0,35,37,37,0

```

## 二十三、 ON 开关语句

开关语句又称为控制转向语句。R1机没有这个语句。

形式：
$$\text{ON } I \left\{ \begin{array}{l} \text{GOTO} \\ \text{GOSUB} \end{array} \right\} n_1, n_2, n_3, \dots, n$$

I是算术表达式或有确切值的变量，但必须大于等于0，不能小于0。

$n_1, n_2, n_3, \dots$ 是行号

当I是1时转向第一个行号的语句，是2时则转向第二个行号的语句。……

当I等于n时转向第n个的行号语句，但I不能大于n。

例：某商店售出各商品如下表，求出各种商品和总销售金额。

货物名	编号	售出量	单价	金额
毛巾	1	102	0.53元	?
牙膏	2	268	0.62元	?
茶杯	3	76	3.25元	?
饭盒	4	42	1.88元	?
录音机	5	65	306.50元	?
⋮	⋮	⋮	⋮	
⋮	⋮	⋮	⋮	

总金额?

设：M——各种货物售出量

G——各种货物金额

H——全部货物金额

编写程序如下：



```

10 INPUT I, M
20 ON I GOSUB 100,200,300,400,500,.....
30 PRINT G
40 H=H+G
50 PRINT H
60 GOTO 10
70 END
100 G=M*0.5
110 RETURN
200 G=M*0.62
210 RETURN
300 G=M*3.25
310 RETURN
400 G=M*1.88
410 RETURN
500 G=M*306.5
510 RETURN
:
:
```

开关语句的另一种用法是在程序中出现意想不到的错误时，使机器停止运行，可用：

ON ERR GOTO n n是行号  
使程序转向n行使计算机继续运行。

## 二十四、逻辑判断

电子计算机中，电子器件只能表示两种状态，开或关，不能有第三种状态。电子计算机采用二进制，用0或1表示。

电子计算机的电路叫数字电路，或叫门电路。

逻辑判断就是根据数学表达式的值或变量(为1或0)来判断，或根据逻辑式来进行判断。

基本的电路有以下三种：

与门——输入端全是1，输出端才是1。用AND表示。

或门——输入端有一个是1，输出端就是1，用OR表示。

非门——输入端是0，输出端与输入端相反。用NOT表示  
运用逻辑判断会给我们编写程序带来很大方便。

形式：IF 关系表达式 THEN 语句1：语句2：语句3

例一：比较A、B、C值的大小，并打印出最大值。

```
10 INPUT A, B, C
20 IF A>B AND A>C THEN PRINT
   A: GOTO 100
30 IF B>A AND B>C THEN PRINT
   B: GOTO 100
40 PRINT C
100 END
```

例二：判断A、B、C三个值，A是否是最小值？

```
10 INPUT A, B, C
20 IF A>B OR A>C THEN PRINT A,
   "NOTMIN": GOTO 40
30 PRINT "MIN="; A
40 END
```

说明：

1. THEN后可以跟复合多重语句，语句间用：号分

开。 注：R1机不可以接复合语句


2. 如条件成立时,要把THEN后的所有语句执行完,如条件不成立,就继续执行这条语句的下一条语句。

## 二十五、循环的转出

循环可以有多个嵌套,在APPLE II型机最多可以嵌套10层循环,但在循环嵌套和转出中应注意以下几点:

在循环嵌套中除内外循环体不得交叉外,在循环语句和条件语句或转向语句联合使用时,可以从循环体内转到循环体外,但不允许从循环体外转入循环体内。多重循环只允许从内循环体转到外循环体,不允许从外循环体转入内循环体。多重循环只能经过FOR向下执行,不能从中间或从后面向前执行。但是从循环体内转到循环体外执行一段,还可以再转到循环体内。

如:下图是正确的



如:下图是不正确的



## 二十六. 彩色作图

彩色作图分为低分辨率作图和高分辨率作图两种。

注：R1机没有彩色作图。

(一). 低分辨率作图语句：

1. GR

形式 GR

执行这条语句, 计算机打开彩色通道, 把屏幕置于低分辨率图形显示方式(40×40), 屏幕下部留出4行用作正文显示。

如果需要变成满屏幕图形显示, (40×48) 就可输入 POKE-16302, 0 或 POKE-49324, 0。再要恢复通常方式可输入 TEXT


2 COLOR=数学表达式

低分辨率彩色作图共16种色彩, 编号是0~15。数学表达式代表色彩号。色彩号如下：

0 黑色	1 品红	2 深兰	3 紫
4 深绿	5 灰	6 天兰	7 淡兰
8 褐	9 橙	10 灰	11 粉红
12 绿	13 黄	14 湖兰	15 白

如不送入这条语句, 就按COLOR=0对待。

3. PLOT 数学式1, 数学式2



数学式1是X坐标, 数学式2是Y坐标, 以此标出画点(色

块)的位置。

X和Y值都要在0~39之间。画点的颜色由离PLOT最近的COLOR决定。同一画点上如给了两种颜色，后给的颜色要取代先给的颜色。

4. HLIN 数学式1, 数学式2 AT 数学式3

数学式1和数学式2是X坐标,分别是X轴上的两点。数学式3是Y坐标。这个语句是要在Y行上,用色块画一水平线。

5. VLIN 数学式1, 数学式2 AT 数学式3

数学式1和数学式2是Y坐标,分别是Y轴上的两点。数学式3是X坐标,这个语句是要在X列上,用色块画一垂直线。

例题: 10 GR  
20 FOR I=0 TO 15  
30 COLOR=I  
40 HLIN 0,39 AT I  
50 VLIN 0,39 AT I  
60 NEXT I  
70 END

## (二) 高分辨率作图

1. HGR HGR2

形式: HGR  
HGR2


执行这条语句,使计算机打开彩色通道,把屏幕置于高分辨率图形显示方式。(HGR是 $280 \times 160$ ,下部留出四行用作正文显示。HGR2是 $280 \times 192$ ,是满屏幕)

2. HCOLOR=数学表达式

高分辨率彩色作图共8种色彩,编号是0~7。各色彩色显示器常有不同的颜色。

3. HPLOPT 数学式1, 数学式2 TO 数学式3, 数学式4

即: n(行号) HPLOT X1, Y1 TO X2, Y2



X1, X2是X坐标, Y1, Y2是Y坐标。在这两点间画一线。  
(垂直、水平、斜线都可)。

X值要在0~279之间

Y值要在0~159之间, 满屏幕显示要在0~191之间。

如果下一条线的起点是这条线终点时, 在同一语句中,  
可以写成: HPLOT X1, Y1 TO X2, Y2 TO X3, Y3 TO X4, Y4

如不同在一语句时, 可以写成:

```
HPLOT X1, Y1 TO X2, Y2
```

```
HPLOT TO X3, Y3
```

```
HPLOT TO X4, Y4
```


例:

```
10 HGR
20 FOR I=0 TO 7
30 HCOLOR=I
40 HPLOT 10,10 TO 269,10 TO 269,149 TO
10,149 TO 10,10
50 NEXT I
```

## 二十七、R1机黑白作图

## (一)画点作图。

形式：PLOT 数学式1,数学式2



数学式1是X坐标，数学式2是Y坐标，以此标出画点(反白)的位置。X值要在0~63之间，Y值要在0~43之间，可以用画点联成图形。

## (二)几何图形作图

利用GRAPHICS键和  等图案编程序作图。

形式 PRINT "GRAPHICS  GRAPHICS"

说明：

1. 按GRAPHICS键后，屏幕上出现一白底黑G字光标表明计算机进入作图状态；要退出作图状态时还要再按一次GRAPHICS键。

2. 按GRAPHICS键还可以作反白指令，按这个键出现白底黑G字光标后，再送入字符都是白底黑字。要再按这个键才能退出反白状态。

3. 图案作图屏幕与正文显示相同，即：

$$32(0\sim 31)\times 23(0\sim 22)$$

## 二十八、R1机的音乐程序

R1机的音乐程序比其它类型机简单易学。

形式：MUSIC “ ”

“ ”中是乐谱代码。

### (一)乐谱简谱代码是

简谱	.....5	6	7	1	2	3	4	5	6
代码	.....G	< A	< B	< C	D	E	F	G	A
	7	1	2	3	.....				
	B	C >	D >	E >					

>是高音符号，<是低音符号。

简谱	1*	2*	4*	5*	6*
代码反白	C	D	F	G	A

### (二)节拍代码是：

一拍	4
二拍	8
半拍	2
休止符	.

### (三)速度指令 TEMPO U

U是0~255之间的数值。数值越大，速度越慢，数值越小，速度越快。不送入TEMPO时，按一般速度。

使用TEMPO U时要放在MUSIC语句前一行语句里。

(四)编制音乐程序时，如没有编上行号，就只演奏一遍，演奏完这一遍后程序就失去作用。如编上行号，可执行多次。如要自动重复执行，可在程序末尾使用GOTO语句，



使程序重新执行。

```
例： 10 PRINT AT 15, 9; "YING HUA"  
      20 FOR U=1 TO 300  
      30 NEXT U  
      40 A$="A4 A4 B8 A4 A4 B8"  
      50 B$="A4 B4 C>4 B4 A4 B2 A2  
F8 E4 C4 E4 F4 E4 E2 C2 B<8"  
      60 C$="E4 F4 B2 A2 F4 E12"  
      70 MUSIC A$  
      80 MUSIC B$  
      90 MUSIC B$  
     100 MUSIC A$  
     110 MUSIC C$  
     120 GOTO 10
```

## 第三章 文件存取和打印机使用

### 一. 文件存取

将主机内存中的程序，存到驱动器磁盘上备用；需用时，再将磁盘上存的程序调入主机使用。文件存取有以下指令：

#### (一) SAVE 文件名

执行这个指令可以把主机内存中的程序，以自定的文件名，存到接通的驱动器中的磁盘上(磁盘需要先进行初始化——指令见后)

文件名必须以英文字母开头，最长不要超过30个字符，除，号外都可做文件名。文件名与SAVE当中要空一格。同一磁盘上不能存以相同的文件名存入程序，否则后存入的将先存入的程序清除。

#### (二) LOAD 文件名

指行这个指令可以把接通的驱动器中磁盘上的，以这一文件名存入的程序，调入主机内存。调入后，即将计算机内原内存程序清除。待驱动器红灯熄灭后，就可以送入LIST或RUN指令。

#### (三) RUN 文件名

不再经LOAD文件名，就可将磁盘外存程序调入主机内存并执行一遍。

#### (四) CATALOG

调入并显示磁盘外存程序目录。在CATALOG后可加D1或D2。D1、D2是驱动器编号。

目录中第一个符号是\*。

第二个字符是文件属性。A是浮点BASIC编写的文件  
I是整形BASIC文件，B是二进制文件。

后面的数字表示文件所占磁盘区段数，表示所占容量的大小。

再后是文件名。

在有)号时，调用B文件要输入BRUN、BLOAD。如要恢复)符时，需输入FP。

(五) DELETE 文件名

删去磁盘上这一文件名的文件程序。

(六) INIT

执行这一个指令可将磁盘初始化。通常输入 INIT 后输入HELLO作为引导程序名

(七) COPY

将一个磁盘上的全部文件复制到另一个磁盘上，不需再初始化。

## 二. 打印机的使用：

步骤是：

(一).上纸。如在没有上纸或纸将用完时接通打印机，就要发出声响，红灯亮，以示警告，这时不可打印。

(二).接通电源开关，指示灯亮。

(三).送入PR#1指令。READY和ONLINE灯亮。

(四).送入LIST指令，可将主机内存程序在打印机上打印出来。

(五).送入RUN指令，将程序运行结果打印出来。

(六). 送入PR # 0指令。使打印机与计算机主机脱机，不能再打印主机内存程序。ONLINE灯灭。

(七). 撕纸或按打印机上LF键，慢速走纸或转动纸轴走纸。尽量不要用FF键快速走纸。切不可向相反方向将纸拉出来。

(八). 关电源开关。

高分辨率作图图形需在图形显示停止不动时方可用打印机打印。打印时需在售入PR # 1以后，再同时按CTRL键和Q键，才能打印出图形。

### 三. 大小写

BASIC语言指令中不能用小写字母。但如要在字符串中使用小写字母时，需使用转换大小写的指令。即同时按SHIFT CTRL和U键，就可输入小写字母。如需恢复输入大写字母时，仍需再按以上三个键。

有的计算机设有大小写转换键。R1机不能输入小写字母。

附图一 APPLE I 型机 键盘图 (紫金 I)

1	"	#	%	&	'	(	)	^	*	[ ]	STOP RUN
1	2	3	4	5	6	7	8	9	:	=	-
ESC	Q	W	E	R	T	Y	U	I	O	P	RET STOP URN LIST
CTRL	A	S	D	F	G	H	J	K	L	+	@REPT ← →
SHIFT	Z	X	C	V	B	N	M	<	>	?	/





## R1型机常用信息表

信 息	说 明
O K	好
S T	执行的最后程序
O M	溢出存储器
U V	未定义变量
S F	显示器屏满(需再显示输入CONT)
A G	某些函数自变量无效
I R	实数在范围之外
N F	NEXT 在 FOR之外
O V	数溢出,计算时导出数字大于 $10^{38}$
R G	RETURN 在 GOSUB以外
B S	下标不对
I I	非法输入
B K	程序被 BREAK 键中断
M F	音乐字符串格式不对
N A	程序定名为空字符串

## APPLE II 机型 错误信息表

(一) 发生了一个错误之后。BASIC 返回到以 ) 提示符指定的命令且显示出闪烁的光标。程序文本和变量值均不破坏, 但程序不能够继续执行了。而且将所有的 GOSUB 和 FOR 循环计数置 0。

(二) 在程序运行过程中为了避免错误引起的中断。可以使用错误处理程序 ONERR GOTO 语句。

(三) 在立即执行方式中发出错误时。不打印行标号。出错误信息的格式是:

立即执行模式                    ? × ×    ERROR

迟缓执行模式                    ? × ×    ERROR IN YY

在上面的两种情况里            “××” 是具体的错误标志  
“YY” 是迟缓执行模式发生错误的行标号。在这种模式下语句中的错误在没执行前不检查。

错误信息	说            明
CAN'T CONTINUS	试图继续执行一个不存在的、或发生了错误后的或从程序中删除了语句行后的程序
DIVISION BY ZERO	除数为 0 出错
ILLEGAL DIRECT	不能使用 INPUT、DEF FN、GEF 和 DATA 语句作为直接执行模式命令
ILLEGAL QUANTITY	对于算术函数或字符串函数传递的参数超出范围、在下面几种情况下可能发生这种类型的错误: a、负的数组下标 (例 LET A(-1)=0) ; b、LOG 函数的自变量是负值或 0 ; c、SQR 函数的自变量是负值; d、A^B A 是负值 B 不是整数 e、在 MID\$ LEFT\$ RIGHT\$ WAIT



	PEEK POKE TAB SIC ON...GOTO 或任何图形函数中使用了不合适的自变量
NEXT WITHOUT FOR	NEXT语句中的变量与仍然有效的FOR语句中的变量不对应, 或一个无名的NEXT对应任何一个仍然有效的FOR
OUT OF DATA	DATA中的数据不够读, 或者DATA中含有不合适的数据
OUT OF MEMORY	上面的任何一种情况都可能产生该类型的错; 误程序太长; 变量太多FOR循环嵌套多于10层; GOSUB的嵌套多于24层; 形式很复杂的表达式。括号嵌套多于36层; 试图置LOMEM, 到太高; 试图置LOMEM 低于现行值; 试图置HIMEM, 太低
FORMULA TOO COMPLEX	为IF "××" THEN执行的命令多于两个语句
REDIM'D ARRAY	在一个数组已经说明了之后, 遇到了对同一个数组的另一个说明语句, 这个错误经常发生的情况是: 已经给定了数组的非特指说明值10象 $A(I) = 3$ 其后在程序中又去说明该数组 DIMA(100)
RETURN WITHOUT GOSUB	执行RETURN语句没有相对应GOSUB语句
STRING TOO LONG	试图利用连续操作建立一个长于255个字符的字符串
BAD SUBSCRIPT	试图使用超出了数组说明的数组元素, 当使用和说明的数组维数不匹配时也会产生这个错误。例: LET A(1,1,1)=Z但A已由DIM A(2,2)说明
SYNTAX ERROR	表达式中的括号失配, 在语句行中的非法字符, 不正确的标点等
TYPE MISMATCH	值语句的左边是算术变量而右边是串、或函数的自变量应是串却给出的是数字, 反之亦然
UNDEF'D STATEMENT	试图GOTO GOSUB或THEN到不存在的语句标号去

UNDEF'D FUNCTION	试图使用没有定义的自定义函数
OVERFLOW	计算的结果大于BASIC提供的的数据形式。 如果发生向下溢出的情况 给出0作为结果并且不打印出错误信息 程序继续执行
WRITE PROTECTED	书写保护

# 习 题

## 第一章

一、求  $1+1+1+1+\cdots+1=?$

100个1

程序如下，请输入计算机

NEW ↵ (注：↵是按回车键符号)

10 LET S=0 ↵

20 FOR I=1 TO 100 STEP 1 ↵

30 LET S=S+1 ↵

40 NEXT I ↵

50 PRINT "S="; S ↵

60 END ↵

RUN ↵

(注：R1机没有END指令，不要输入，下同)

二、一百块正方形水泥板，最小一块边长1米，每块依次增加一米，问这一百块水泥板的面积总和是多少？

即  $1 \times 1 + 2 \times 2 + 3 \times 3 + \cdots + 100 \times 100 = ?$

程序如下，请输入计算机。

NEW ↵

10 LET S=0 ↵

20 FOR X=1 TO 100 ↵

30 LET S=S+X\*X ↵

40 NEXT X ↵

```

50 PRINT "S=" ; S↵
60 END↵
   RUN↵

```

(注：R1机第30句应改为LET S=S+I\*\*2)。

三、将下列程序输入，看显示出什么图形。

```

NEW↵
10 FOR I=1 TO 10↵
20 PRINT TAB (15 I); ↵
30 FOR J=I TO 2 * I-1↵
40 PRINT " * "; ↵
50 NEXT J↵
60 PRINT↵
70 NEXT I↵
80 END↵
   RUN↵

```

如将这个程序中第40句删去；或改为，或看程序执行结果有什么变化？

四、将下列程序输入R1机。

```

NEW↵
10 MUSIC "G2 G2 G4 G2 G2 G4 E2
   G2 A2 G1 E1 D2 E2 G4 G2 E2
   G2 E1 D1 C2 E2 D4 E2 E2 D4"↵
20 MUSIC "A<2 C2 D4 A4 A2 G2 E2 A2
   G4 G2 E2 D2 E2 G8 G2 E2 D2 E2
   G2 E2 D2 E2 A<2 C2 D2 E2 C8"↵
30 GOTO 10↵
   RUN↵

```

## 第二章一至十一节

五、把下列各数字写成科学记数法：

1) 8923764

2) 0.00000000003

六、把下列各数写成日常计数法：

1) 8.76876E+8

2) 0.654321E-6

七、把下列数学式用BASIC语言表示：

1) 
$$\frac{80 - 16 \times 15}{16 + 57 \div 3}$$

2) 
$$A \cdot B \cdot \frac{e1 \cdot e2}{d^2}$$

八、设 $X=8$ ,  $Y=4.5$ , 如想在程序运行后打印出下面结果, 程序应如何编写? 请输入计算机检验。

$$X * Y = 8 * 4.5 = 36$$

九、写出下列程序, 输入计算机进行计算：

1)  $10^4 - 10^2$

2) 圆半径 $r=10$ , 求圆面积

十、写出以下程序执行结果, 输入计算机检验。

```
10 PRINT "A", "B", "C", "D", "E",
```

```
20 PRINT 2*2, 3*3, 4*4, 5*5, 6*6, -2,  
-3, -4
```

```
30 PRINT "X", "Y", "Z"
```

```
40 PRINT 0.2, 0.3, 0.4, -5, -6, -7, -3
```

```
50 END
```

十一、写出打印结果进行比较，输入计算机检验。

1) PRINT G\*M/2\*R

2) PRINT "G\*M/2\*R"

十二、编程序输入计算机用\*号打印一个任选字。

十三、用字符串变量输入计算机打印汉语拼音“爱科学”。

十四、改正下列语句中的错误：

1) 10 LET A=7, B=8, C=9

2) 10 LET N+4=M+8

3) 10 LET P^2=4

注R1机：P\*\*2=4

十五、写出下列程序执行结果，输入计算机检验。

1) 10 A=15

20 B=23

30 C=A

40 A=B

50 B=C

60 PRINT "A=" ; A

70 PRINT "B=" ; B

80 END

2) 10 A=15

20 B=23

30 C=A

40 D=B

50 A=D

60 B=C

70 PRINT "A=" ; A



## 第二章十二至十六节

十八、下列语句有什么错误：

1) 10 FOR I=3 TO 5 STEP -1

2) 10 FOR I=1 TO 5 STEP 0.5

⋮

50 NEXT X

3) 10 FOR I=1 TO 4

20 FOR J=1 TO 5

30 PRINT J, I

40 NEXT I

50 NEXT J

60 END

4) 10 TAB(10), "\*"

20 TAB(15), "\*"

30 TAB(20), "\*"

40 END

十九、写出下列程序执行结果并输入计算机检验。

1) 10 FOR X=4 TO 8 STEP 1

20 PRINT X,

30 NEXT X

40 END

2) 10 FOR X=4.5 TO 8 STEP 2.5

20 PRINT X

30 NEXT X

40 END



```

3) 10 FOR Y=-5 TO -15 STEP -1
    20 LET X=Y+1
    30 PRINT X,
    40 NEXT Y
    50 END

4) 10 X=2
    20 FOR Y=5 TO 10 STEP 2
    30 Z=Y*X
    40 PRINT Z
    50 NEXT Y
    60 END

```

二十、下列程序循环几次？结果是什么？输入计算机检验。

```

1) 10 FOR J=15 TO 6 STEP 2
    20 PRINT J
    30 NEXT J
    40 END

2) 10 FOR J=4 TO 12 STEP -1
    20 PRINT J
    30 NEXT J
    40 END

```

二十一、编制程序用循环语句求1到100间奇数和，并输入计算机。

二十二、编制程序用循环语句求1到100间偶数和，并输入计算机。

二十三、打印出1到10之间各奇数的平方与各偶数的立方。

二十四、打印出以下图形（第一个\*号在第10列）

1) \* \* \* \* \*                      2)                      \* \* \* \* \*

     \* \* \* \* \*    \* \* \* \* \*

     \* \* \* \* \*    \* \* \* \* \*

     \* \* \* \* \*    \* \* \* \* \*

3)                      \* \*                      4)                      \*

                         \*                      \*    \*                      \*

                         \*    \*    \*    \*

                         \*    \*    \*    \*

5)                      \*                      6)    \*

                         \*                      \*    \* \* \*

                         \*    \*    \* \* \* \* \*

                         \*    \*    \* \* \* \*

                         \*                      \*    \*

                         \*    \* \* \* \* \*

二十五、打印九九乘法表。

二十六、用循环语句编程序计算  $\frac{A}{A^2 - 1}$  的值。

其中  $A = 5 \sim 15$  STEP 5

二十七、某厂在试制新产品时需计算下列两式：

$$X = \frac{(R^2 + 6.3R)^3 - R}{2} \qquad Y = \frac{3.45X}{1.1X - 3}$$

其中要求R从2.6~3.7每隔0.2计算一点。

二十八、第十七题栏目中增加一栏“参加人数”，设每

校参加50人。

二十九、自选一个表格题。

三十、计算  $Y = \frac{X^2}{1!} + \frac{X^3}{2!} + \frac{X^3}{3!} + \frac{X^5}{4!} + \dots + \frac{X^{N+1}}{N!}$

设  $X=2$ ，请打印出  $N$  分别等于 1、2、3、4、5、6、7、8、9、10 时的  $Y$  值。

三十一、用循环语句编写“加法器”（用来计算几个数之和）的程序。用 INPUT 语句从键盘送入若干个数据（每次送入数据个数和数值可以不同，请自定）。要求打印出数据的个数、数的总和。

三十二、小明想用“\*”号打印两条交叉的斜线，他编制了以下这个程序，但没有出现他予想的结果，请想一想这是什么原因？

```
10 FOR A=1 TO 21
20 PRINT TAB(5+A); "*" ;TAB (26 -
A); "*"
30 NEXT A
40 END
```

## 第二章 第十七节至二十节

三十三、下列程序有什么错误

- 1) 10 IF  $X \geq 0$  THEN 80
- 2) 10 IF  $X + Y \neq 3 * 5$  THEN 50
- 3) 10 INPUT X  
20. IF  $X < 0$  THEN 100  
30. PRINT X  
40. END

三十四、编程序计算 $X=3.5$   $X=-4.8$ 时 $Z$ 的值

$$Z = \begin{cases} \frac{2X^2 + 3X + 1}{4X^2 + 4X + 2} & X > 0 \\ \frac{2X^2 + 3X + 1}{X^2 + 1} & X \leq 0 \end{cases}$$

三十五、编一程序，输入计算机，判断一组数中的数是否大于100，打印出结果（编程序前要画框图）。

三十六、编一程序可以比较两数的大小（编程序前要画框图）。

三十七、三个正方形边长分别是2、4、6、……60，求各自正方形周长（编程序前要画框图）。

三十八、输入十个数，打印出其中最大数。

三十九、输入五个数，把最大的数和它的顺序号打印出来。

四十、输入十个数，统计出其中正数、负数和零的个数。

四十一、一个数列，它的头三个数是0，0，1。第四

个数是前三个数之和。以后每个数都分别是它前三个数之和。请编一程序，打印出这个数列。直到第六十个数或数的值达到（或超过） $10^{15}$ 为止。

四十二、有一半径10厘米的圆板，上有半径3厘米、2厘米、1厘米的圆孔，求这圆板完整时面积和三个孔的面积各多少？圆板面积是多少？（用子程序）

四十三、打印出以下五个组各三个数，并列出每组中最大数（用子程序）。

- 1) 3, 19, 16
- 2) 12.6, 7.8, 9.1
- 3) -5, -1, -7
- 4) 215, 309, 316
- 5) 0, -1, -2

四十四、下列程序运行结果是什么？

- 1) 

```
10 READ A, B, C, D, E, F
20 T=A+B+C+D+E
30 U=T/F
40 PRINT "U=" ; U
50 DATA 8, 4
60 DATA 6, 15, 4, 5
70 END
```
  
- 2) 

```
10 READ A, B
20 DATA 1, 2, 3
30 READ C, D, E
40 RESTORE
50 DATA 4, 5, 6, 7
```

```
60 READ F, G, H, I, J, K, L, M, N
70 RESTORE
80 DATA 8, 9, 10

90 PRINT A, B, C, D, E, F, G, H,
      I, J, K, L, M, N

100 END
```

四十五、编制一个把华氏温度变成摄氏温度的程序。  
F是华氏温度值，C是摄氏温度值。公式是：

$$C = \frac{5}{9} (F - 32)$$

求当华氏 $F=32, 60, 48, 82, 98.6$ 时摄氏值

## 第二章 第二十一节至二十二节

四十六、

1) 第一个角的SIN、第二个角的COS、第三个角的TAN, 三个角分别是 $30^\circ$ 、 $60^\circ$ 、 $45^\circ$ , 请打印出函数值。

2) 当ASN、ACS、ATN的函数值分别是0.5、0.5、1时, 在R1机上打印出它们的度数。

四十七、宇宙飞船脱离地球引力影响(第二宇宙速度), 速度必须等于或大于 $V = \sqrt{2gR}$ , 问V最小要达到多少?

( $R=6371$ 公里  $g=9.8$ )

四十八、设 $y = ((\cos X + \operatorname{tg} X) \cdot (\ln 5 + \sqrt{X^2 + 2}))$ 求y值(键盘输入X值)。

四十九、指出以下各式中的X值的范围

1)  $\operatorname{INT}(X+5)=17$

2)  $\operatorname{INT}(100 * X + 0.5) / 100 = 8.91$

3)  $\operatorname{INT}(10 * X + 2) / 10 = 7.2$

五十、写出以下范围的随机数

1) 0~10以内的

2) 0~100以内的

3) 30~40

五十一、给计算机输入十个数, 打印出其中奇数、偶数及零的个数

五十二、编制程序求两个正整数的最小公倍数。

五十三、编制程序打印出3~100间的所有的素数(只能被1和自己本身整除的数是素数)

五十四、編程序打印

$$A = \begin{vmatrix} 4 & 5 & 6 \\ 7 & 8 & 9 \end{vmatrix}$$

五十五、編制程序能把若干个(如15、20、30、……)任意给的数,按大小顺序排列打印出来。

五十六、为小学生编写十道,百位数以内,数不能重复的加减法算术题。



## 第二章 第二十三节至二十五节

五十七、某商店有十种商品，每种商品又有3~4个品种，各售出数量如下，求各种商品售出数和总售出量。

种类\品种	1	2	3	4	合计
一	28	32	64	79	?
二	21	0	9	5	?
三	3	7	9	18	?
四	78	96	105	26	?
五	6	0	0	62	?
六	15	76	94	9	?
七	21	76	32	64	?
八	7	0	9	5	?
九	3	7	9	4	?
十	128	96	42	0	?
合计					?

五十八、有八根火柴，由计算机和小朋友轮流拿1根、2根，最多不能超过3根，也不许不拿，谁拿到最后一根谁就是胜利者。请小朋友先拿。编一个程序，使计算机始终胜利。

## 第二章 第二十六节至二十八节 第三章

五十九、分别用低分辨率、高分辨率作图或写美术字。  
R1机可用画点或图案作图，写美术字。

六十、用R1机编制一音乐程序。

六十一、练习文件存取。

六十二、练习打印机的使用。

## 习题参考答案

五、1)  $8.923764E+6$       2.  $3E-10$

六、1) 876876000      2. 0.000000654321

七、1)  $(80-16*15)/(16-57/3)$

2)  $X=A*B*E1*E2/D^2$

R1机第2小题改为 $X=A*B*E1*E2/D**2$

八、10 X=8

20 Y=4.5

30 PRINT "X\*Y=";X;"\*";Y;"=";X\*Y

40 END

九、1)10 LET A=10^4      2) 10 LET R=10

20 LET B=10^3

20 S=3.14159\*R^2

30 PRINT A-B

30 PRINT "S=";S

40 END

40 ENL

RUN

RUN

9000

S=3.14159

R1机第10句改成

R1机第20句改成

LET A=10\*\*4

S=3.14159\*R\*\*2

20句改成B=10\*\*3

	A	B	C	R1机	A	B
十、	D	E			C	D
	4	9	16		E	
	25	36	-2		4	9
	-3	-4			16	25

X	Y	Z	36	-2
0.2	0.3	0.4	-3	-4
-5	-6	-7	X	Y
-8			Z	
			0.2	0.3
			0.4	-5
			-6	-7
			-8	

十一、1) 0, R1机出现UV      2)  $G * M / 2 * R$

十三、10 A\$ = "AI KE XUE"

20 PRINT A\$

30 END

十四、1) 10 A=7: B=8: C=6

R1机: 10 LET A=7

20 LET B=8

30 LET C=9

2)、3)等号左边不能是表达式

十五、1) A=23

B=15

2) A=23

B=15

```

十六、 10 PRINT " * * * * * * * * * "
        20 PRINT " * "
        30 PRINT " * "
        40 PRINT " * "
        50 PRINT " * "
        60 PRINT " * "
        70 PRINT " * "
        80 PRINT " * "
        90 PRINT " * "
        100 PRINT " * "
        110 PRINT " * "
        120 PRINT " * "
        130 PRINT " * "
        140 PRINT " * "
        150 PRINT " * "
        160 PRINT " * "
        170 PRINT " * "
        180 PRINT " * "
        190 PRINT " * "
        200 PRINT " * "
        210 PRINT " * "
        220 END

```

```

十七、 10 PRINT " * JI SUAN JI JING SAI
        YOU XIU TONG JI * "
        20 PRINT
        30 PRINT "....."
        40 PRINT
        50 PRINT "XIAOMING", "1SHI", "2SHI"

```

```
60 PRINT
70 PRINT "....."
80 A$ = "2 ZHONG"
90 B$ = "8 ZHONG"
100 C$ = "3 XIAO"
110 D$ = "ZHONG XIN XIAO XUE"
120 E$ = "AI KE XIAO XUE"
130 A1=20
140 B1=19
150 C1=15
160 D1=15
170 E1=16
180 A2=11
190 B2=9
200 C2=10
210 D2=9
220 E2=11
230 PRINT
240 PRINT A$, A1, A2
250 PRINT
260 PRINT B$, B1, B2
270 PRINT
280 PRINT C$, C1, C2
290 PRINT
300 PRINT D$, D1, D2
310 PRINT
320 PRINT E$, E1, E2
```

```

330 PRINT
340 PRINT "....."
350 X1=A1+B1+C1+D1+E1
360 X2=A2+B2+C2+D2+E2
370 PRINT
380 PRINT "HE JI" , X1, X2
390 END

```

注：R1机，50语句改为PRINT "XIAO MING" .  
"YOU XIU"

180~220语句取消。240~320语句中A2 B2 C2 D2  
D2 E2及前面一个，号取消。360语句取消。380语句中X2  
及前一个，号取消。

- 十八、1) 初值、终值、步长之间关系不正确。  
2) FOR后变量与NEXT后变量不一致。  
3) 双重循环交叉。  
4) 行号后TAB前缺PRINT。，号应为；号。

十九、1) 4      5      6            2) 4.5  
      7      8                        7  
3) -4      -5      -6      4) 10  
      -7      -8      -9            14  
      -10     -11     -12           18  
      -13     -14

二十、1) 一次。15            2) 一次。4

```

二十一、10 X=0
          20 FOR I=1 TO 100 STEP 2
          30 X=X+I
          40 NEXT I

```

```
50 PRINT "X=" ; X
```

```
60 END
```

二十二、10 X=0

```
20 FOR I=1 TO 100 STEP 2
```

```
30 X=X+I+1
```

```
40 NEXT I
```

```
50 PRINT "X=" ; X
```

```
60 END
```

二十三、10 FOR X=1 TO 10 STEP 2

```
20 Y=X+1
```

```
30 PRINT X^2,
```

```
40 PRINT Y^3
```

```
50 NEXT X
```

```
60 END
```

二十四、1)

```
10 FOR I=1 TO 4
```

```
20 FOR J=1 TO 5
```

```
30 PRINT TAB(10+I-1) ; "*" ;
```

```
40 NEXT J
```

```
50 PRINT
```

```
60 NEXT I
```

```
70 END
```

2)10 FOR I=1 TO 4

```
20 FOR J=1 TO 5
```

```
30 PRINT TAB(10-I+1) ; "*" ;
```

```
40 NEXT J
```

```
50 PRINT
```



```

60 NEXT I
70 END
3)10 FOR I=1 TO 4
    20 PRINT TAB(10-I+1); "*" ;
        TAB(10+I); "*"
    30 NEXT I
    40 END
4)10 PRINT TAB(10); "*"
    20 FOR I=1 TO 3
    30 PRINT TAB(10-I); "*" ;
    40 PRINT TAB(10+I); "*"
    50 NEXT I
    60 END
5)10 PRINT TAB(10); "*"
    20 FOR I=1 TO 3
    30 PRINT TAB(10-I); "*" ;
    40 PRINT TAB(10+I); "*"
    50 NEXT I
    60 FOR I=3 TO 2 STEP -1
    70 PRINT TAB(10-I+1); "*" ;
    80 PRINT TAB(10+I-1); "*"
    90 NEXT I
    100 PRINT TAB(10); "*"
    110 END
6)答案 (一)
    10 FOR I=1 TO 3
    20 FOR J=1 TO I*2-1

```

```
30 PRINT TAB(10-I+1); "*" ;
40 NEXT J
50 PRINT
60 NEXT I
70 FOR I=2 TO 1 STEP -1
80 FOR J=1 TO 2*I-1
90 PRINT TAB(10-I+1); "*" ;
100 NEXT J
110 PRINT
120 NEXT I
130 END
```

答案 (二)

```
10 FOR I=0 TO 2
20 FOR J=-I TO I
30 PRINT TAB(10-I); "*" ;
40 NEXT J
50 PRINT
60 NEXT I
70 FOR I=1 TO 0 STEP -1
80 FOR J=-I TO I
90 PRINT TAB(10-I); "*" ;
100 NEXT J
110 PRINT
120 NEXT I
130 END
```

二十五、答案 (一)

```
10 FOR I=1 TO 9
```

```

20 FOR J=1 TO 9
30 PRINT I; " * "; J; " = "; I * J; " "
40 NEXT J
50 PRINT
60 NEXT I
70 END

```

答案 (二)

```

10 FOR I=1 TO 9
20 FOR J=1 TO I
30 PRINT I; " × "; J; " = "; I * J; " "
40 NEXT J
50 PRINT
60 NEXT I
70 END

```

二十六、

```

10 FOR A=5 TO 15 STEP 5
20 PRINT "R=" ; A/(A^2-1)
30 NEXT A
40 END

```

注：R1机第20语句改为PRINT "R=" ; A/(A \*\* 2-1)

二十七、

```

10 FOR R=2.6 TO 3.7 STEP 0.2
20 X=(R^2+6.3*R)^3-R)/2
30 Y=(3.45*X)/(1.1*X-3)
40 PRINT "R=" ; R
50 PRINT "X=" ; X

```

```

60 PRINT "Y=" ; Y
70 NEXT R
80 END

```

二十八 第十七题答案作如下修改:

```

50 PRINT TAB(1); "XIAO MING" ;
  TAB(18); "1 SHI" ; TAB(25);
  "2 SHI" ; TAB (33); "REM SHU"
225 R=50
240 PRINT TAB(1); A$ ; TAB(18); A1;
  TAB(25); A2; TAB(33); R
260 PRINT TAB(1);B$ ; TAB(18);B1;
  TAB (25);B 2; TAB(33);R
280 PRINT TAB(1);C$ ; TAB(18);C1;
  TAB (25);C2;TAB(33);R
300 PRINT TAB(1);D$ ; TAB(18);D1;
  TAB (25);D2;TAB(33);R
320 PRINT TAB(1);E$ ; TAB(18);E1;
  TAB (25);E2; TAB(33);R
380 PRINT TAB(1);"HEJI"; TAB(18);X1;
  TAB(25);X2; TAB(33);R*5

```

注: R1机程序略

三十 答案1

```

10 Y=0
20 X=2
30 B=1
40 FOR N=1 TO 10
50 A=X*X\N

```

答案2

```

10 Y=0
20 X=2
30 A=X
40 B=1
50 FOR N=1 TO 10

```

60	B=B*N	60	A=A*X
70	S=A/B	70	B=B*N
80	Y=Y+S	80	Y=Y+A/B
90	PRINT "N=";	90	PRINT "N=";N,
	N, "Y="; Y		"Y="; Y
100	NEXT N	100	NEXT N
110	END	110	END

三十一、

```

10 INPUT "SHU JU SHU=";N
20 S=0
30 FOR I=1 TO N STEP 1
40 INPUT "SHU JU=";X
50 S=S+X
60 NEXT I
70 PRINT "SHU ZONG HE=";S, "SHU
JU SHU=";N
80 END

```

三十二. 因在执行过程中,  $(5+A)$  的值会逐渐增加  
' 大于  $(20-A)$ 。

三十三、1)  $\geq$  错误, 应为  $>$  2)  $\neq$  错误, 应为  $<>$   
3) 20 语句中 THEN 100, 但整个程序最大行  
号是 40, 没有 100

```


三十四、10 INPUT X
20 IF X>0 THEN 60
30 Z=(2*X^2+3*X+1)/(X^2+1)
40 PRINT "X="; X, "Z="; Z
50 GOTO 10

```


4 \* X + 2)

```
60 Z=(2 * X^2+3 * X+1) / (4 * X^3+
70 PRINT "X=";X, "Z=";Z
80 GOTO 10
90 END
```

三十五



```
10 INPUT X
20 IF X > 100 THEN 50
30 PRINT "X < 100 X =", X
40 GOTO 60
50 PRINT "X > 100 X =", X
60 GOTO 10
70 END
```



```


10 PRINT "2 SHU BI DA
    XIAO"
20 INPUT A, B
30 PRINT
40 IF A < B THEN 70
50 IF A > B THEN 90
60 GOTO 110
70 PRINT A; "<"; B
80 COTO 120
90 PRINT A; ">"; B
100 GOTO 120
110 PRINT A; "="; B
120 GOTO 10
130 END
  
```

注：R1机第20句改为  
另加

```

20 INPUT A
25 INPUT B
  
```

### 三十七



```

10 L=2
20 S=4*L
30 PRINT "L=" ; L
40 PRINT "S=" ; S
50 IF L=60 THEN 80
60 L=L+2
70 GOTO 20
80 END
  
```

### 三十八

```

10 INPUT A
20 FOR I=2 TO 10
30 INPUT B
40 IF B<A THEN
   60
50 A=B
60 NEXT I
70 PRINT "ZUI DA
   SHU=" ; A
80 END
  
```

送入第一数  
 循环九次  
 送入第二个数, 共九个数  
 大数存在 A 所在单元里  
 打印最大数



三十九.	10	R=1	第一个顺序号赋给 R
	20	INPUT X	送入第一个数
	30	FOR I=2 TO 5	循环4次
	40	INPUT Y	送入第二个数, 共
	50	IF Y<X THEN 70	4个数
	60	X=Y	大数存在X所在单 元里
	70	R=I	大数顺序号存在
	80	NEXT I	所在单元里
	90	PRINT "X=";X	打印最大数
	100	PRINT "R=";R	打印最大数顺序号
	110	END	

四十.	10	Z=0	正数累计个数 先置0
	20	F=0	负数累计个数 先置0
	30	L=0	零数累计个数 先置0
	40	FOR I=1 TO 10	
	50	INPUT X	
	60	IF X<=0 THEN 90	
	70	Z=Z+1	
	80	GOTO 130	
	90	IF X=0 THEN 120	
	100	F=F+1	
	110	GOTO 130	
	120	L=L+1	
	130	NEXT I	

```
140 PRINT "Z=";Z,"F=";F,"L=";L
150 END
```

#### 四十一

```
10 INPUT A,B,C
20 FOR I=1 TO 20
30 PRINT A, B, C
40 A=A+B+C
50 IF A>10E+15 THEN 110
60 B=B+C+A
70 IF B>10E+15 THEN 110
80 C=C+A+B
90 IF C>10E+15 THEN 110
100 NEXT I
110 END
```

注：R1机第10句改为

```
5 INPUT A
10 INPUT B
15 INPUT C
```

四十二 X是圆板面积 S1是完整的圆板面积  
S2 S3 S4 是三个孔面积

```
10 R=10
20 GOSUB 500
30 PRINT "S1=";S
40 X=S
50 R=3
60 GOSUB 500
70 PRINT "S2=";S
80 X=X-S
```

```

90 R=2
100 GOSUB 500
110 PRINT "S3=";S
120 X=X-S
130 R=1
140 GOSUB 500
150 PRINT"S4=";S
160 X=X-S
170 PRINT "X=";X
180 END
500 S= π * R ^ 2
510 RETURN

```

注：R1机第500语句改成 $S = \pi * R * * 2$

#### 四十三

10 INPUT "A=";A	注：R1机程序改为
20 INPUT"B=";B	10 INPUT A
30 INPUT "C=";C	20 INPUT B
40 GOSUB 500	30 INPUT C
50 GOTO 10	40 GOSUB 300
60 END	50 GOSUB 500
500 IF A>B	60 GOTO 10
THEN 520	
510 A=B	300 PRINT "A=";A
520 IF A>C THEN 550	310 PRINT "B=";B
530 PRINT C	320 PRINT "C=";C
540 RETURN	330 RETURN
550 PRINT A	500 以下各语句同

## 四十四

1.  $U=7.4$
2.   1  2  3  
      4  5  1  
      2  3  4  
      5  6  7  
      8  9

## 四十五.

```

10 PRINT "F", "C"
20 READ F
30 C=(5/9)*(F-32)
40 PKINT F, C
50 GOTO 20
60 DATA 32,60,48,82,98.6
70 END

```

## 四十六.

```

1)10 PRINT SIN(30*3.14159/180)
   20 PRINT COS(60*3.14159/180)
   30 PRINT TAN(45*3.14159/180)
   RUN
   0.5
   0.5
   1

```

注：R1机     3.14159改为  $\pi$

2) R1机

```

10 PRINT ASN 0.5/( $\pi$ /180)

```

```

20 PRINT ACS 0.5/( $\pi/180$ )
30 PRINT ATN 1/( $\pi/180$ )
RUN
30
60
45

```

四十七.

```

10 R=6371*1000
20 G=9.8
30 V=SQR(2*R*G)
40 PRINT "V="; V
50 END
RUN
11174.596

```

四十八.

```

10 INPUT "X="; X
20 Y=ABS((COS(X)+TAN(X)*(LOG(5)+
SQR(X^2+2)))
30 PRINT "Y="; Y
40 END

```

四十九.

- 1) 12—12.9999999
- 2) 8.91—8.914
- 3) 7.01—7.09

五十

- |               |              |
|---------------|--------------|
| 1) RND(1)*10  | R1机: RND*10  |
| 2) RND(5)*100 | R1机: RND*10) |

3)  $RND(S) * 10 + 30$  R1机:  $RND * 10 + 30$

### 五十一

```
10 J=0                J是累计奇数的个数
20 E=0                E是累计偶数的个数
30 L=0                L是累计零数的个数
40 FOR I=1 TO 10
50 INPUT X
60 IF X=0 THEN 120    先判断是零吗?
70 IF INT(X/2)=X/2 THEN 100 再判断是偶
                        数吗?
80 J=J+1
90 GOTO 130
100 E=E+1
110 GOTO 130
120 L=L+1
130 NEXT I
140 PRINT "J=", J, "E=", E, "L="; L
150 END
```

### 五十二

```
10 INPUT A, B
20 I=1
30 X=A*I
40 Y=INT(X/B)*B
50 IF X=Y THEN 80
60 I=I+1
70 GOTO 30
80 PRINT A, B, X      X是最小公倍数
```

90 END

注：R1机第10语句改为INPUT A  
增加15 INPUT B

### 五十三

```
10 FOR X=3 TO 100
20 FOR I=2 TO M-1      定下除数范围
30 IF INT(X/I)=X/I THEN 60
40 NEXT I
50 PRINT X
60 NEXT X
70 END
```

### 五十四

```
10 DIM A(2,3)
20 FOR I=1 TO 2
30 FOR J=1 TO 3
40 READ A(I,J)
50 PRINT A(I,J);
60 NEXT J
70 PRINT
80 NEXT I
90 DATA 4,5,6,7,8,9
100 END
```

### 五十五

```
10 INPUT N
20 DIM A(N)
30 FOR I=1 TO N
40 READ A(I)
```

```

50 NEXT I
60 FOR I=1 TO N-1
70 FOR X=1+I TO N
80 IF A(I)>A(X) THEN 120
90 B=A(I)
100 A(I)=A(X)
110 A(X)=B
120 NEXT X
130 NEXT I
140 FOR I=1 TO N
150 PRINT A(I)
160 NEXT I
170 DATA N个数
180 END

```

五十六.

```

10 S=0 : F=0
20 FOR I=1 TO 10
30 A=INT(RND(1)*100)
40 B=INT(RND(1)*100)
50 READ E
60 IF E=10 THEN 110
70 PRINT A;"+"; B;"=";
80 INPUT C
90 IF A+B<>C THEN 170
100 GOTO 150
110 IF A<B THEN GOSUB 500
120 PRINT A;"-";B;"=";

```



```

130 INPUT C
140 IF A-B<>C THEN 170
150 S=S+10
160 PRINT "10";"GOOD";GOTO 190
170 F=F+(-10)
180 PRINT "-10";"WRONG"
190 NEXT I
200 DATA 10,10,2,2,10,10,3,3,10,10
210 END
300 T=A : A=B : B=T
510 RETURN

```

### 五十七

```

10 DIM A(10,4)
20 FOR I=1 TO 10
30 G=0
35 FOR J=1 TO 4
40 READ A(I,J)
50 G=G+A(I,J)
60 NEXT J
70 PRINT G
80 W=W+G
90 NEXT I
100 PRINT TAB(15);W
110 DATA 28,32,..... (40个数)
120 END

```

### 五十八

```

10 INPUT X1

```

```
20 IF X1>3 OR X1<0 THEN 570
30 ON X1 GOTO 100,200,300
100 J1=3
105 GOTO 310
200 J1=2
210 GOTO 310
300 J1=1
310 S1=8-J1-X1
320 PRINT "X1=";X1,"J1=";J1,"S1=";S1
330 INPUT X2
340 IF X2>3 OR X2<0 THEN 570
350 ON X2 GOTO 400,450,500
400 J2=3
410 GOTO 550
450 J2=2
460 GOTO 550
500 J2=1
550 S2=4-J2-X2
560 PRINT "X2=";X2,"J2=";J2,"S2=";S2
570 END
```